

# **EXCEL Formulas**

## **Excel 365**

**Prepared & presented by**

**Mubarak Ghaith**

**September 2023**

# Table of Contents

1. SUM of Digits when cell Contains all Numbers .....	2
2. SUM of Digits when cell Contains Numbers and non Numbers both .....	2
3. A List is Unique or Not (Whether it has duplicates) .....	2
4. Count No. of Unique Values .....	2
5. Count No. of Unique Values Conditionally .....	2
6. Generate Sequential Weekday names like Sun, Mon, Tue, .....,Sat .....	3
7. Generate Sequential Month names like Jan, Feb, Mar....Dec.....	4
8. Find Last Day of the Month .....	4
9. Number of Days in a Month .....	4
10. Find First Day of the Month.....	4
11. Add Month to or Subtract Month from a Given Date.....	5
12. Add Year to or Subtract Year from a Given Date.....	5
13. Convert a Number to a Month Name .....	5
14. Convert a Month Name to Number .....	6
15. Convert a Number to Weekday Name.....	6
16. Convert a Weekday Name to Number.....	6
17. Financial Year Formula (e.g. 2015-16 or FY16) .....	7
18. Converting Date to a Calendar Quarter .....	7
19. Converting Date to a Indian Financial Year Quarter.....	7
20. Determine Quarter for Fiscal Year .....	8
21. Calculate Age from Given Birthday .....	8
22. Convert from dd/mm/yy to mm/dd/yy (DMY to MDY) .....	8
23. Convert from mm/dd/yy to dd/mm/yy (MDY to DMY) .....	9
24. Number to Date Format Conversion.....	9
25. Number to Time Format Conversion.....	10
26. Convert Time to Decimal Hours, Minutes and Seconds .....	10
27. Convert Decimal Hours, Minutes and Seconds to Time .....	10
28. Generate a Sequence of Dates .....	11
29. Generate a Sequence of Times .....	11
30. How to Know if a Year is a Leap Year .....	11
31. Last Working Day of the Month If a Date is Given .....	12
32. First Working Day of the Month if a Date is Given .....	12
33. Determine Number of Working Days in a Year .....	13
34. Determine Number of Working Days in a Month .....	13
35. How Many Mondays or any other Day of the Week between 2 Dates.....	14

36.	Find Number of Friday the 13th between Two Given Dates .....	14
37.	Calculate Next Working day if date falls on a Weekend / Holiday.....	14
38.	Calculate Previous Working day if date falls on a Weekend / Holiday .....	15
39.	Date for Nth Day of the Year .....	16
40.	Extract Date and Time from Date Timestamp .....	16
41.	First Working Day of the Year .....	16
42.	Last Working Day of the Year .....	17
43.	Convert from Excel Date (Gregorian Date) to Julian Date.....	17
44.	Convert from Julian Dates to Excel (Gregorian) Dates .....	18
45.	Convert a Number into Years and Months .....	18
46.	Find the Next Week of the Day.....	18
47.	Find the Previous Week of the Day .....	19
48.	Count Cells Starting (or Ending) with a particular String .....	20
49.	Count No. of Cells Having Numbers Only .....	20
50.	Reverse a String .....	20
51.	Reverse Number String.....	20
52.	Reverse a String of Words.....	21
53.	Reverse a Range of Cells .....	21
54.	Get First Row Number in a Range.....	21
55.	Get Last Row Number in a Range.....	22
56.	Get First Column Number in a Range .....	22
57.	Get Last Column Number in a Range .....	22
58.	Extract Unique Characters.....	22
59.	Test whether a range contains only numbers.....	22
60.	Test whether a range contains only English alphabets.....	22
61.	Count No. of Cells which are containing only Characters.....	23
62.	Number of Characters in a String without considering blanks .....	23
63.	Number of times a character appears in a string.....	23
64.	Count Non Numbers in a String.....	23
65.	Count Numbers in a String.....	24
66.	Count only Alphabets in a String.....	24
67.	Generate English Alphabets .....	24
68.	Convert English Alphabets to Numbers .....	25
69.	Convert Numbers to English Alphabets .....	25
70.	Extract nth Word from Front.....	25
71.	Extract nth Word from Back .....	25
72.	Extract first two words OR first n words .....	26
73.	Extract last two words OR last n words .....	26

74.	Most Frequently Occurring Value in a Range.....	27
75.	COUNTIF with OR Condition.....	27
76.	SUMIF with OR Condition .....	28
77.	COUNTIF on Filtered List .....	28
78.	SUMIF on Filtered List.....	28
79.	Extract First Name from Full Name .....	28
80.	Extract Last Name from Full Name .....	28
81.	Extract the Initial of Middle Name.....	28
82.	Extract Middle Name from Full Name.....	29
83.	Remove Middle Name in Full Name .....	29
84.	Extract Integer and Decimal Portion of a Number .....	29
85.	Maximum Times a Particular Entry Appears Consecutively.....	30
86.	Get File Name through Formula .....	30
87.	Get Workbook Name through Formula .....	30
88.	Get Sheet Name through Formula .....	30
89.	Get Workbook's Directory from Formula.....	31
90.	Perform Multi Column VLOOKUP .....	31
91.	VLOOKUP from Right to Left.....	32
92.	Case Sensitive VLOOKUP .....	33
93.	Rank within the Groups .....	33
94.	Extract Alphabets from a String .....	34
95.	Remove Alphabets from a String.....	34
96.	Extract numbers from string .....	34
97.	Remove numbers from string.....	34
98.	Roman Representation of Numbers.....	35
99.	Sum Bottom N Values in a Range .....	35
100.	Sum Every Nth Row .....	36
101.	Sum Every Even Row.....	36
102.	Sum Every Odd Row .....	36
103.	Sum Top N values in a Range .....	37
104.	We have AVERAGEIF. What about MEDIANIF and MODEIF?.....	37
105.	Calculate Geometric Mean by Ignoring 0 and Negative Values.....	38
106.	Generate GL Codes.....	39
107.	Abbreviate Given Names.....	39
108.	Get Column Name for a Column Number .....	40
109.	Get Column Range for a Column Number.....	40
110.	Find the nth Largest Number when there are duplicates .....	41
111.	COUNTIF for non-contiguous range .....	41

112.	Count the Number of Words in a Cell / Range.....	42
113.	Numerology Sum of the Digits aka Sum the Digits till the result is a single digit ...	42
114.	Generate Sequential Numbers and Repeat them.....	43
115.	Repeat a Number and Increment and Repeat.....	43
116.	Generate Non Repeating Random Numbers through Formula.....	44
117.	Extract User Name from an E Mail ID .....	44
118.	Extract Domain Name from an E Mail ID .....	45
119.	Location of First Number in a String .....	45
120.	Location of Last Number in a String .....	45
121.	Find the Value of First Non Blank Cell in a Range .....	45
122.	Find First Numeric Value in a Range .....	45
123.	Find Last Numeric Value in a Range .....	45
124.	Find First non Numeric Value in a Range .....	45
125.	Find Last non Numeric Value in a Range .....	46
126.	Find Last Used Value in a Range .....	46
127.	I have data for many years but I want the sum for only last 12 months .....	46
128.	Generate a Unique List out of Duplicate Entries.....	47
129.	Financial Function - Calculate EMI.....	47
130.	Financial Function - Calculate Interest Part of an EMI.....	48
131.	Financial Function - Calculate Principal Part of an EMI.....	50
132.	Financial Function - Calculate Number of EMIs to Pay Up a Loan.....	51
133.	Financial Function - Calculate Interest Rate.....	52
134.	Financial Function – Calculate Compounded Interest.....	54
135.	Financial Function – Calculate Effective Interest .....	55
136.	Financial Function – Calculate CAGR and AAGR.....	56
137.	Slab Billing – Calculate Income Tax, Electricity (Utility) Bills based on Slabs .....	57
138.	LTRIM and RTRIM through Excel Formulas.....	58

## **A Note about Dynamic Arrays (not for Excel 365 / Excel 2021)**

For Excel versions which don't have Dynamic Arrays functionality, non Dynamic Arrays equivalents are also given immediately after Dynamic Arrays formulas.

## **A Note about Array formulas (not for Excel 365 / Excel 2021)**

Sometimes, you will need to enter a formula as array formula. In Excel 365/Excel 2021, all formulas are treated as Array formula, hence you need not enter any formula as Array formula. Only for older versions of Excel, you might need to enter a formula as Array formula.

Array Formula is not entered by pressing ENTER after entering your formula but by pressing CTRL+SHIFT+ENTER. If you are copying and pasting this formula, take F2 after pasting and CTRL+SHIFT+ENTER. This will put { } brackets around the formula which you can see in Formula Bar. If you edit again, you will have to do CTRL+SHIFT+ENTER again. Don't put { } manually.

## 1. SUM of Digits when cell Contains all Numbers

If you cell contains only numbers like A1:= 7654045, then following formula can be used to find sum of digits

```
=SUM(--MID(A1,SEQUENCE(LEN(A1)),1))
```

```
=SUMPRODUCT(--MID(A1,ROW(INDIRECT("1:"&LEN(A1))),1))
```

```
=SUM(--MID(A1,ROW(INDIRECT("1:"&LEN(A1))),1))
```

If A1 is blank, then to handle error, you can enclose above formulas into an IFERROR block.

## 2. SUM of Digits when cell Contains Numbers and non Numbers both

If your cell contains non numbers apart from numbers like A1:= 76\$5a4b045%d, then following formulas can be used to find sum of digits

```
=SUM(IFERROR(--MID(A1,SEQUENCE(LEN(A1)),1),0))
```

```
=SUMPRODUCT((LEN(A1)-LEN(SUBSTITUTE(A1,ROW($1:$9),"")))*ROW($1:$9))
```

```
=SUM(IFERROR(--MID(A1,ROW(INDIRECT("1:"&LEN(A1))),1),0))
```

## 3. A List is Unique or Not (Whether it has duplicates)

Assuming, your list is in A1 to A1000. Use following formula to know if list is unique.

```
=MAX(COUNTIF(A1:A1000,A1:A1000))
```

If answer is 1, then it is Unique. If answer is more than 1, it is not unique.

## 4. Count No. of Unique Values

Use following formula to count no. of unique values -

```
=IF(COUNTA(A1:A100)=0,0,COUNTA(UNIQUE(FILTER(A1:A100&"",A1:A100<>""))))
```

```
=SUMPRODUCT((A1:A100<>"")/COUNTIF(A1:A100,A1:A100&""))
```

```
=SUM((A1:A100<>"")/COUNTIF(A1:A100,A1:A100&""))
```

## 5. Count No. of Unique Values Conditionally

If you have data like below and you want to find the unique count for Region = "A", then you can use below formula -

=IF(COUNTIF(A2:A20,"A")=0,0,COUNTA(UNIQUE(FILTER(B2:B20,A2:A20="A"))))

=SUM(IF(FREQUENCY(IF(A2:A20<>"",IF(A2:A20="A",MATCH(B2:B20,B2:B20,0))),ROW(A2:A20)-ROW(A2)+1),1))

If you have more number of conditions, the same can be built after A2:A20 = "A".

	A	B
1	<b>Region</b>	<b>Sales Man</b>
2	A	a
3	A	a
4	A	b
5	A	b
6	A	b
7	A	c
8	A	c
9	A	a
10	B	m
11	B	n
12	B	o
13	B	p
14	B	d
15	B	d
16	B	e
17	B	e

## 6. Generate Sequential Weekday names like Sun, Mon, Tue, .....,Sat

For 3 characters Weekday name

=TEXT(DATE(2017,1,SEQUENCE(7)),"ddd")

=TEXT(DATE(2017,1,ROW(\$1:\$7)),"ddd")

(Note – Year 2017 has been used as 1-Jan-2017 was Sunday. You can use any year in place of 2017 where 1-Jan was a Sunday like year 2023)

**For full Weekday name, replace ddd with dddd**

=TEXT(DATE(2017,1,SEQUENCE(7)),"dddd")

=TEXT(DATE(2017,1,ROW(\$1:\$7)),"dddd")

You can add a number if you want to show some other Weekday Name. Say, if you want to show 1 = Monday, 2 = Tuesday.....7 = Sunday, just add 1 . Hence, formulas would become

=TEXT(1+DATE(2017,1,SEQUENCE(7)),"ddd")

=TEXT(1+DATE(2017,1,ROW(\$1:\$7)),"ddd")



The above formulas will generate the sequence in a column. If you want to generate in a row, use following formulas

```
=TEXT(DATE(2017,1,SEQUENCE(7)),"ddd")
```

```
=TRANSPOSE(TEXT(DATE(2017,1,ROW($1:$7)),"ddd"))
```

## 7. Generate Sequential Month names like Jan, Feb, Mar....Dec

Generate 3 characters month names

```
=TEXT(DATE(1,SEQUENCE(12),1),"mmm")
```

```
=TEXT(DATE(1,ROW($1:$12),1),"mmm")
```

For full month name, replace mmm with mmmm

```
=TEXT(DATE(1,SEQUENCE(12),1),"mmmm")
```

```
=TEXT(DATE(1,ROW($1:$12),1),"mmmm")
```

The above formulas will generate the months names in a column. To generate in a row –

```
=TEXT(DATE(1,SEQUENCE(,12),1),"mmm")
```

```
=TRANSPOSE(TEXT(DATE(1,ROW($1:$12),1),"mmm"))
```

## 8. Find Last Day of the Month

Suppose, you have a date in the cell A1:= 14-Aug-22, then formula for finding last day of the month is

```
=EOMONTH(A1,0)
```

## 9. Number of Days in a Month

Suppose, you have been given a date say 15-Nov-21 and you have to determine how many days this particular month contains.

The formula which you need to use in the above case would be

```
=DAY(EOMONTH(A1,0))
```

## 10. Find First Day of the Month

Suppose, you have a date in the cell A1:= 14-Aug-21, then formula for finding first day of the month is

=A1-DAY(A1)+1

=EOMONTH(A1,-1)+1

=DATE(YEAR(A1),MONTH(A1),1)

## 11. Add Month to or Subtract Month from a Given Date

Very often, you will have business problems where you have to add or subtract month from a given date. One scenario is calculation for EMI Date.

Say, you have a date of 10/22/21 (MM/DD/YY) in A1 and you want to add number of months which is contained in Cell B1.

The formula in this case would be

=EDATE(A1,B1)

[Secondary formula =DATE(YEAR(A1),MONTH(A1)+B1,DAY(A1)) ]

Now, you want to subtract month which is contained in Cell B1.

=EDATE(A1,-B1)

[Secondary formula =DATE(YEAR(A1),MONTH(A1)-B1,DAY(A1)) ]

## 12. Add Year to or Subtract Year from a Given Date

In many business problems, you might encounter situations where you will need to add or subtract years from a given date.

Let's say A1 contains Date and B1 contains numbers of years.

If you want to add Years to a given date, formulas would be -

=EDATE(A1,12\*B1)

=DATE(YEAR(A1)+B1,MONTH(A1),DAY(A1))

If you want to subtract Years from a given date, formulas would be -

=EDATE(A1,-12\*B1)

=DATE(YEAR(A1)-B1,MONTH(A1),DAY(A1))

## 13. Convert a Number to a Month Name

Use below formula to generate named 3 lettered month like Jan, Feb....Dec

```
=TEXT(A1*30,"mmm")
```

Replace "mmm" with "mmmm" to generate full name of the month like January, February....December in any of the formulas in this post.

## 14. Convert a Month Name to Number

Say Cell A1 contains the string January, February....December (or Jan. Feb.....Dec) and you want to show 1, 2.....12

```
=MONTH("1"&A1)
```

The formula would work as long as month names are >=3 characters. Hence, it would work for say Janu or Decem or Apri or Octobe.

## 15. Convert a Number to Weekday Name

Suppose you want to return 1 = Sunday, 2 = Monday.....7 = Saturday

```
=TEXT(DATE(2017,1,A1),"dddd")
```

Note – 2017 has been used in above formula as 1-Jan-2017 was Sunday. You can use any year where 1-Jan was Sunday like year 2023.

To show only 3 characters of the Weekday Name, replace dddd with ddd

```
=TEXT(DATE(2017,1,A1),"ddd")
```

You can add a number to A1 if you want to show some other Weekday Name

Say, if you want to show 1 = Monday, 2 = Tuesday.....7 = Sunday, just add 1 to A1

```
=TEXT(1+DATE(2017,1,A1),"dddd")
```

Say, if you want to show 1 = Friday, 2 = Saturday.....7 = Thursday, just add 5 to A1

```
=TEXT(5+DATE(2017,1,A1),"dddd")
```

## 16. Convert a Weekday Name to Number

Say Cell A1 contains the string Sunday, Monday....Saturday (or Sun, Mon.....Sat) and you want to show 1, 2.....7, then following formula can be used to return the numbers. Sunday will be 1 and Saturday will be 7.

```
=ROUND(SEARCH(LEFT(A1,2),"SuMoTuWeThFrSa")/2,0)
```

```
=MATCH(LEFT(A1,2),{"Su","Mo","Tu","We","Th","Fr","Sa"},0)
```

If we want to return some other number to weekdays, then formula can be tweaked accordingly. For example, to make Mon = 1 and Sun = 7

```
=ROUND(SEARCH(LEFT(A1,2),"MoTuWeThFrSaSu")/2,0)
```

```
=MATCH(LEFT(A1,2),{"Mo","Tu","We","Th","Fr","Sa","Su"},0)
```

## 17. Financial Year Formula (e.g. 2015-16 or FY16)

A good number of countries don't follow calendar year as the financial year. For example, India's financial year start is 1-Apr and finishes on 31-Mar. Hence, currently (20-Feb-16), the financial year is 2015-16 (It is also written as FY16). On 1-Apr-16, it will become 2016-17 (It is also written as FY17).

Now if a date is given, then following formula can be used to derive 2015-16 kind of result.

```
=YEAR(A1)-(MONTH(A1)<=3)&"-"&YEAR(A1)+(MONTH(A1)>3)
```

To generate FY16 kind of result, following formula can be used

```
="FY"&RIGHT(YEAR(A1)+(MONTH(A1)>3),2)
```

## 18. Converting Date to a Calendar Quarter

Assuming date is in Cell A1. You want to convert it into a quarter (1, 2, 3 & 4). Jan to Mar is 1, Apr to Jun is 2, Jul to Sep is 3 and Oct to Dec is 4.

```
=CEILING(MONTH(A1)/3,1)
```

OR

```
=ROUNDUP(MONTH(A1)/3,0)
```

OR

```
=CHOOSE(MONTH(A1),1,1,1,2,2,2,3,3,3,4,4,4)
```

## 19. Converting Date to a Indian Financial Year Quarter

Assuming date is in Cell A1. You want to convert it into a Indian Financial Year Quarter. Jan to Mar is 4, Apr to Jun is 1, Jul to Sep is 2 and Oct to Dec is 3.

```
=CEILING(MONTH(EDATE(A1,-3))/3,1)
```

OR

```
=ROUNDUP(MONTH(EDATE(A1,-3))/3,0)
```

OR

```
=CHOOSE(MONTH(A1),4,4,4,1,1,1,2,2,2,3,3,3)
```

## 20. Determine Quarter for Fiscal Year

Few countries follow different quarter other than Q1 from Jan-Mar and Q2 for Apr-Jun. In case of Jan-Mar as Q1, formula is simple (if cell A2 is date)

```
=ROUNDUP(MONTH(A2)/3,0)
```

This will give result as 1, 2, 3 & 4 for the quarters. If you want, you can concatenate "Q" in the formula to show Q1, Q2 etc as below

```
="Q"&ROUNDUP(MONTH(A2)/3,0)
```

If your financial / fiscal year starts in Apr, then for Jan-Mar, quarter is 4 whereas for Apr to Jun, quarter is 1 and so on. In this case, you can use following formula

```
=CEILING(MONTH(EDATE(A1,-3))/3,1)
```

```
= ROUNDUP(MONTH(EDATE(A1,-3))/3,0)
```

If your financial / fiscal year starts in Jul, then for Jan-Mar, quarter is 3 whereas for Jul to Sep, quarter is 1 and so on. In this case, you can use following formula

```
=CEILING(MONTH(EDATE(A1,-6))/3,1)
```

```
= ROUNDUP(MONTH(EDATE(A1,-6))/3,0)
```

If your financial / fiscal year starts in Oct, then for Jan-Mar, quarter is 2 whereas for Oct to Dec, quarter is 1 and so on. In this case, you can use following formula

```
=CEILING(MONTH(EDATE(A1,-9))/3,1)
```

```
= ROUNDUP(MONTH(EDATE(A1,-9))/3,0)
```

## 21. Calculate Age from Given Birthday

```
=DATEDIF(A1,TODAY(),"y")&" Years "&DATEDIF(A1,TODAY(),"ym")&" Months  
&DATEDIF(A1,TODAY(),"md")&" Days"
```

## 22. Convert from dd/mm/yy to mm/dd/yy (DMY to MDY)

Say you have following dates in DMY format

```
24/8/22  
24/8/2022  
4/08/92  
04/08/1992
```

And you need to convert them into MDY format, then use the following formula

**Case1** – if your default date format is MDY

```
=FILTERXML("<t><s>"&SUBSTITUTE(TEXT(A1,"mm/dd/yyyy"),"/","</s><s>")&"</s></t>>","//s[2]")&"/"&FILTERXML("<t><s>"&SUBSTITUTE(TEXT(A1,"mm/dd/yyyy"),"/","</s><s>")&"</s></t>","//s[1]")&"/"&FILTERXML("<t><s>"&SUBSTITUTE(TEXT(A1,"mm/dd/yyyy"),"/","</s><s>")&"</s></t>","//s[3]")
```

**Case2** – if your default date format is DMY

```
=FILTERXML("<t><s>"&SUBSTITUTE(TEXT(A1,"dd/mm/yyyy"),"/","</s><s>")&"</s></t>>","//s[2]")&"/"&FILTERXML("<t><s>"&SUBSTITUTE(TEXT(A1,"dd/mm/yyyy"),"/","</s><s>")&"</s></t>","//s[1]")&"/"&FILTERXML("<t><s>"&SUBSTITUTE(TEXT(A1,"dd/mm/yyyy"),"/","</s><s>")&"</s></t>","//s[3]")
```

## 23. Convert from mm/dd/yy to dd/mm/yy (MDY to DMY)

Say you have following dates in MDY format

8/24/22

8/24/2022

8/04/92

08/04/1992

And you need to convert them into DMY format, then use following formula

**Case1** – if your default date format is MDY

```
=(FILTERXML("<t><s>"&SUBSTITUTE(TEXT(A1,"mm/dd/yyyy"),"/","</s><s>")&"</s></t>>","//s[2]")&"/"&FILTERXML("<t><s>"&SUBSTITUTE(TEXT(A1,"mm/dd/yyyy"),"/","</s><s>")&"</s></t>","//s[1]")&"/"&FILTERXML("<t><s>"&SUBSTITUTE(TEXT(A1,"mm/dd/yyyy"),"/","</s><s>")&"</s></t>","//s[3]"))
```

**Case2** – if your default date format is DMY

```
=FILTERXML("<t><s>"&SUBSTITUTE(TEXT(A1,"dd/mm/yyyy"),"/","</s><s>")&"</s></t>>","//s[2]")&"/"&FILTERXML("<t><s>"&SUBSTITUTE(TEXT(A1,"dd/mm/yyyy"),"/","</s><s>")&"</s></t>","//s[1]")&"/"&FILTERXML("<t><s>"&SUBSTITUTE(TEXT(A1,"dd/mm/yyyy"),"/","</s><s>")&"</s></t>","//s[3]")
```

**Note** – Basically MDY to DMY and DMY to MDY are same formulas as we are just swapping the place of MM with DD and doing nothing else.

## 24. Number to Date Format Conversion

If you have numbers like 010216 and you want to convert this to date format, then the following formula can be used

```
=--TEXT(A1,"00\00\00") for 2 digits year
```

**Note** – Minimum 5 digits are needed for above formula to work

If you have numbers like 01022016 and you want to convert this to date format, then the following formula can be used

=--TEXT(A1,"00\ /00\ /0000") for 4 digits year

Note – Minimum 7 digits are needed for above formula to work

## 25. Number to Time Format Conversion

If you have numbers like 1215 and you want to convert this to hh:mm format, then the following formula can be used

=--TEXT(A1,"00\ :00")

Note – Minimum 3 digits are needed for above formula to work

To convert to hh:mm:ss format

=--TEXT(A1,"00\ :00\ :00")

Note – Minimum 5 digits are needed for above formula to work

## 26. Convert Time to Decimal Hours, Minutes and Seconds

Say A1 has the time 11:35 PM, then you can use following formulas

To convert into Decimal hours

=A1\*24

To convert into Decimal minutes

=A1\*1440

(1440 is nothing but 24\*60)

To convert into Decimal seconds

=A1\*86400

(86400 is nothing but 24\*60\*60)

**(Note – You will need to format your result cell in decimal format)**

## 27. Convert Decimal Hours, Minutes and Seconds to Time

Say A1 has decimal hours 23.58, then you can use following formula to convert it back into time

=A1/24

If it has decimal minutes say 1415, then you can use following formula to convert it back into time

=A1/1440

If it has decimal seconds say 84900, then you can use following formula to convert it back into time

=A1/86400

**(Note – You will need to format your result cell in Time format)**

## 28. Generate a Sequence of Dates

Generate 90 sequential dates starting 1-Apr-21. Let's say that the date is in cell A1. You can use either of following formulas

=SEQUENCE(90,,A1)

=ROW(INDIRECT(A1&":"&A1+89))

Now, let's generate all dates of a given month. Let's say this is Feb-2021. You can use following formula where A1 has the date 1-Feb-2021

=SEQUENCE(DAY(EOMONTH(A1,0)),,A1)

=ROW(INDIRECT(A1&":"&EOMONTH(A1,0)))

Above formulas will generate dates in a column. To generate in a row

=SEQUENCE(,90,A1)

=TRANSPOSE(ROW(INDIRECT(A1&":"&A1+89)))

=SEQUENCE(,DAY(EOMONTH(A1,0)),A1)

=TRANSPOSE(ROW(INDIRECT(A1&":"&EOMONTH(A1,0))))

## 29. Generate a Sequence of Times

Generate 40 sequential times starting at 11 AM with an increment of 15 minutes where A1:=11:00 AM

=A1+SEQUENCE(40,,15/(24\*60))

=A1+(ROW(1:40)-1)\*15/(24\*60)

## 30. How to Know if a Year is a Leap Year



Let's say that A1 contains the year. To know whether it is a Leap Year or not, use following formula -

```
=MONTH(DATE(A1,2,29))=2
```

```
=DAY(EOMONTH(DATE(A1,2,1),0))=29
```

TRUE means that it is Leap Year and FALSE means that this is not a Leap Year.

## 31. Last Working Day of the Month If a Date is Given

If A1 holds a date, the formula for calculating last Working Day of the month would be

```
=WORKDAY(EOMONTH(A1,0)+1,-1)
```

The above formula assumes that your weekends are Saturday and Sunday.

But, if your weekends are different (e.g. in gulf countries), you can use following formula -

```
=WORKDAY.INTL(EOMONTH(A1,0)+1,-1,"0000110")
```

Where 0000110 is a 7 character string, 1 represents a weekend and 0 is a working day. First digit is Monday and last digit is Sunday. The above example is for Gulf countries where Friday and Saturday are weekends.

You also have an option to give a range which has holidays. In that case, your formula would become

```
=WORKDAY(EOMONTH(A1,0)+1,-1,D1:D10)
```

```
=WORKDAY.INTL(EOMONTH(A1,0)+1,-1,"0000110",D1:D10)
```

Where range D1:D10 contains the list of holidays.

## 32. First Working Day of the Month if a Date is Given

If A1 contains a date, then formula for First Working Day of the month would be

```
=WORKDAY(EOMONTH(A1,-1),1)
```

The above formula assumes that your weekends are Saturday and Sunday.

But, if your weekends are different (e.g. in gulf countries), you can use following formula -

```
=WORKDAY.INTL(EOMONTH(A1,-1),1,"0000110")
```

Where 0000110 is a 7 character string, 1 represents a weekend and 0 is a working day. First digit is Monday and last digit is Sunday. The above example is for Gulf countries where Friday and Saturday are weekends.

You also have option to give a range which has holidays. In that case, your formula would become

```
=WORKDAY(EOMONTH(A1,-1),1,D1:D10)
```

```
=WORKDAY.INTL(EOMONTH(A1,-1),1,"0000110",D1:D10)
```

Where range D1:D10 contains the list of holidays.

### 33. Determine Number of Working Days in a Year

Suppose, you have been given a year in A1 (Say A1 = 2021) and you need to determine the number of working days in a Year, then your formula to determine number of working days would be –

```
=NETWORKDAYS(DATE(A1,1,1),DATE(A1,12,31))
```

The above formula is based on the fact that Saturdays and Sundays are weekends. Starting Excel 2010, you can control the weekends in the formula and function is NETWORKDAYS.INTL

```
=NETWORKDAYS.INTL(DATE(A1,1,1),DATE(A1,12,31),"0000110")
```

In the string "000110" – First digit is Monday and last digit is Sunday. 1 defines that particular day as weekend.

If you have got your list of holidays in a range say B1:B20 (B1:B20 should contain dates in date format), you can have following formulas

```
=NETWORKDAYS(DATE(A1,1,1),DATE(A1,12,31),B1:B20)
```

```
=NETWORKDAYS.INTL(DATE(A1,1,1),DATE(A1,12,31),"0000110",B1:B20)
```

### 34. Determine Number of Working Days in a Month

Let's say you need to find number of working days in year 2022 for the month of Jul where year is in A1 and month in number format is in A2 (Hence A2 should be 7 not Jul), then you can use following formula

```
=NETWORKDAYS(DATE(A1,A2,1),EOMONTH(DATE(A1,A2,1),0))
```

The above formula is based on the fact that Saturdays and Sundays are weekends. Starting Excel 2010, you can control the weekends in the formula and function is NETWORKDAYS.INTL

```
=NETWORKDAYS.INTL(DATE(A1,A2,1),EOMONTH(DATE(A1,A2,1),0),"0000110")
```

In the string "000110" – First digit is Monday and last digit is Sunday. 1 defines that particular day as weekend.

If you have got your list of holidays in a range say B1:B20 (B1:B20 should contain dates in date format), you can have following formulas

```
=NETWORKDAYS(DATE(A1,A2,1), EOMONTH(DATE(A1,A2,1),0),B1:B20)
```

```
=NETWORKDAYS.INTL(DATE(A1,A2,1), EOMONTH(DATE(A1,A2,1),0),"0000110",B1:B20)
```

## 35. How Many Mondays or any other Day of the Week between 2 Dates

Suppose A1 = 23-Jan-16 and A2 = 10-Nov-16. To find number of Mondays between these two dates

```
=SUM(--(WEEKDAY(ROW(INDIRECT(A1&":"&A2)))=2))
```

```
=SUMPRODUCT(--(TEXT(ROW(INDIRECT(A1&":"&A2)),"ddd")="Mon"))
```

```
=SUMPRODUCT(--(WEEKDAY(ROW(INDIRECT(A1&":"&A2)))=2))
```

```
=SUMPRODUCT(--(TEXT(ROW(INDIRECT(A1&":"&A2)),"ddd")="Mon"))
```

"Mon" can be replaced with any other day of the week as per need.

## 36. Find Number of Friday the 13th between Two Given Dates

Assume you have been given two dates

A1:=1-Jan-2014

A2:=25-Nov-2016

You can calculate number of Friday the 13<sup>th</sup> between these two dates by following formula

```
=SUMPRODUCT((WEEKDAY(SEQUENCE(A2-A1+1,,A1))=6)*(DAY(SEQUENCE(A2-A1+1,,A1))=13))
```

```
=SUMPRODUCT((WEEKDAY(ROW(INDIRECT(A1&":"&A2)))=6)*(DAY(ROW(INDIRECT(A1&":"&A2)))=13))
```

## 37. Calculate Next Working day if date falls on a Weekend / Holiday

Suppose you are given a date and you are asked to calculate next working day if date is of weekend. If date is a regular workday, then you should show the same date. For example – 8-Mar-19 is a working day. Hence, you should show the same date. But if this is either 9-Mar-19 or 10-Mar-19 which are Saturday and Sunday, then you must show 11-Mar-19 as the next workday. In this case, formula to be used would be

```
=WORKDAY(A2-1,1)
```

Assuming, your holidays are in E2:E3, then formula would be

=WORKDAY(A2-1,1,\$E\$2:\$E\$3)

	A	B	C	D	E
1	<b>Date</b>	<b>Day</b>	<b>Next Workday</b>		<b>Holidays</b>
2	3/1/2019	Fri	3/1/2019		3/11/2019
3	3/2/2019	Sat	3/4/2019		3/15/2019
4	3/3/2019	Sun	3/4/2019		
5	3/4/2019	Mon	3/4/2019		
6	3/5/2019	Tue	3/5/2019		
7	3/6/2019	Wed	3/6/2019		
8	3/7/2019	Thu	3/7/2019		
9	3/8/2019	Fri	3/8/2019		
10	3/9/2019	Sat	3/12/2019		
11	3/10/2019	Sun	3/12/2019		
12	3/11/2019	Mon	3/12/2019		
13	3/12/2019	Tue	3/12/2019		
14	3/13/2019	Wed	3/13/2019		
15	3/14/2019	Thu	3/14/2019		
16	3/15/2019	Fri	3/18/2019		
17	3/16/2019	Sat	3/18/2019		
18	3/17/2019	Sun	3/18/2019		
19	3/18/2019	Mon	3/18/2019		
20	3/19/2019	Tue	3/19/2019		
21	3/20/2019	Wed	3/20/2019		

Note – If you are using weekends other than Saturday and Sunday, use WORKDAY.INTL with appropriate parameters.

## 38. Calculate Previous Working day if date falls on a Weekend / Holiday

Suppose you are given a date and you are asked to calculate Previous Working day if date is of weekend. If date is a regular workday, then you should show the same date.

For example – 8-Mar-19 is a working day. Hence, you should show the same date. But if this is either 9-Mar-19 or 10-Mar-19 which are Saturday and Sunday, then you must show 8-Mar-19 as the previous workday.

	A	B	C	D	E
1	<b>Date</b>	<b>Day</b>	<b>Previous Workday</b>		<b>Holidays</b>
2	3/8/2019	Fri	3/8/2019		3/15/2019
3	3/9/2019	Sat	3/8/2019		3/18/2019
4	3/10/2019	Sun	3/8/2019		
5	3/11/2019	Mon	3/11/2019		
6	3/12/2019	Tue	3/12/2019		
7	3/13/2019	Wed	3/13/2019		
8	3/14/2019	Thu	3/14/2019		
9	3/15/2019	Fri	3/14/2019		
10	3/16/2019	Sat	3/14/2019		
11	3/17/2019	Sun	3/14/2019		
12	3/18/2019	Mon	3/14/2019		
13	3/19/2019	Tue	3/19/2019		
14	3/20/2019	Wed	3/20/2019		
15	3/21/2019	Thu	3/21/2019		
16	3/22/2019	Fri	3/22/2019		
17	3/23/2019	Sat	3/22/2019		
18	3/24/2019	Sun	3/22/2019		

In this case, formula to be used would be

=WORKDAY(A2+1,-1)

Assuming, your holidays are in E2:E3, then formula would be

=WORKDAY(A2+1,-1,\$E\$2:\$E\$3)

Note – If you are using weekends other than Saturday and Sunday, use WORKDAY.INTL with appropriate parameters.

## 39. Date for Nth Day of the Year

Suppose A1 contains the Year and you are asked to find 69th day of the year which is contained in A2. Then formula for finding Nth day of the year would be

```
=DATE(A1,1,1)+A2-1
```

## 40. Extract Date and Time from Date Timestamp

Suppose you have a date timestamp value in cell A1

A1 = 06/14/15 10:15 PM

And you want to extract date and time out of this.

To extract date, use following formula and format the result cell as date

```
= INT(A1)
```

To extract time, use following formula and format the result cell as time

```
= MOD(A1,1)
```

## 41. First Working Day of the Year

If a year is given in A1 say 2016, below formula can be used to know the first working day of the year (format the result as date)

```
=WORKDAY(DATE(A1-1,12,31),1)
```

```
=WORKDAY(DATE(A1,1,1)-1,1)
```

The above formula assumes that your weekends are Saturday and Sunday.

But, if your weekends are different (e.g. in gulf countries), you can use following formula -

```
=WORKDAY.INTL(DATE(A1-1,12,31),1,"0000110")
```

Where 0000110 is a 7 character string, 1 represents a weekend and 0 is a working day. First digit is Monday and last digit is Sunday. The above example is for Gulf countries where Friday and Saturday are weekends.

You also have option to give a range which has holidays. In that case, your formula would become

```
=WORKDAY(DATE(A1-1,12,31),1,D1:D10)
```

```
=WORKDAY.INTL(DATE(A1-1,12,31),1,"0000110",D1:D10)
```

Where range D1:D10 contains the list of holidays.

## 42. Last Working Day of the Year

If a year is given in A1 say 2016, below formula can be used to know the last working day of the year (format the result as date)

```
=WORKDAY(DATE(A1+1,1,1),-1)
```

```
=WORKDAY(DATE(A1,12,31)+1,-1)
```

The above formula assumes that your weekends are Saturday and Sunday.

But, if your weekends are different (e.g. in gulf countries), you can use following formula -

```
=WORKDAY.INTL(DATE(A1+1,1,1),-1,"0000110")
```

Where 0000110 is a 7 character string, 1 represents a weekend and 0 is a working day. First digit is Monday and last digit is Sunday. The above example is for Gulf countries where Friday and Saturday are weekends.

You also have option to give a range which has holidays. In that case, your formula would become

```
=WORKDAY(DATE(A1+1,1,1),-1,D1:D10)
```

```
=WORKDAY.INTL(DATE(A1+1,1,1),-1,"0000110",D1:D10)
```

Where range D1:D10 contains the list of holidays.

## 43. Convert from Excel Date (Gregorian Date) to Julian Date

**Q.** First what is a Julian Date?

**A.** A Julian date has either 7 digits or 5 digits date and these are generally used in old IT legacy systems.

7 Digits - YYYYDDD - 2016092 (This is 1-Apr-2016. 92 means that this is 92nd day from 1-Jan in that year)

5 Digits - YYDDD - 16092

**Q.** What formulas to use to convert Excel Dates to Julian Dates where A1 has the date say 14-Feb-2022?

**A.** For 7 Digits, use following formula

```
=--TEXT(A1,"yyyy")&TEXT(A1-DATE(YEAR(A1),1,1)+1,"000")
```

For 5 Digits, use following formula

```
=--TEXT(A1,"yy")&TEXT(A1-DATE(YEAR(A1),1,1)+1,"000")
```

## 44. Convert from Julian Dates to Excel (Gregorian) Dates

For 7 Digits Julian Dates, following formula should be used

```
=DATE(LEFT(A1,4),1,RIGHT(A1,3))
```

For 5 Digits Julian Dates, following formula should be used

```
=DATE(YEAR("1-1-"&LEFT(A1,2)),1,RIGHT(A1,3))
```

## 45. Convert a Number into Years and Months

Suppose, you have been given a number into cell A1 say 26 and you want to display it as 2 Years and 4 Months, you can use following formula -

```
=INT(A1/12)&" Years and "&MOD(A1,12)&" Months"
```

Now, an user can become more demanding and he can say that if month is less than 12, then Years should not be displayed. For example, he might say that 8 should be converted to 8 Months and it should not be shown as 0 Years and 8 Months.

In this case, the formula would be -

```
=IF(INT(A1/12)>0,INT(A1/12)&" Years and ","")&MOD(A1,12)&" Months"
```

Now 8 will be displayed as 8 Months only not as 0 Years and 8 Months.

Now, user can ask more. He can say when I give 12, it displays as 1 Years and 0 Months and he simply wants to see 1 Years only. And for 36, he wants to see only 3 Years not 3 Years 0 Months. In this case, formula will have to be tweaked more. Now, the formula becomes -

```
=IF(INT(A1/12)>0,INT(A1/12)&" Years ", "")&IF(MOD(A1,12)=0, "", MOD(A1,12)&" Months")
```

Now an user can come and can ask for one last thing. He can say that if this is 1 Year or 1 Month, it should not be displayed as Years or Months as 1 is not plural. Hence, 25 should be displayed as 2 Years and 1 Month not as 2 Years and 1 Months. Hence, 18 should not be displayed as 1 Years and 6 Months but as 1 Year and 6 Months. Similarly 13 should be displayed as 1 Year and 1 Month not as 1 Years and 1 Months.

```
=IF(INT(A1/12)>0,INT(A1/12)&" Year"&IF(INT(A1/12)>1,"s","")&" and ", "")&MOD(A1,12)&" Month"&IF(MOD(A1,12)>1,"s","")
```

## 46. Find the Next Week of the Day

There are 2 scenarios in this case. For example, if today's date is 2-Jan-17 (Monday) and I try to find the next Monday, I can get either 2-Jan-17 or 9-Jan-17 as per need. For Tuesday to Sunday, it is not a problem as they come after 2-Jan-17 only.

**Case 1 - If the Day falls on the same date, then that very date** (Hence, in case of 2-Jan-17, next Monday would be 2-Jan-17 only)

<b>Next Mon</b>	=CEILING(\$A\$1-2,7)+2
<b>Next Tue</b>	=CEILING(\$A\$1-3,7)+3
<b>Next Wed</b>	=CEILING(\$A\$1-4,7)+4
<b>Next Thu</b>	=CEILING(\$A\$1-5,7)+5
<b>Next Fri</b>	=CEILING(\$A\$1-6,7)+6
<b>Next Sat</b>	=CEILING(\$A\$1-7,7)+7
<b>Next Sun</b>	=CEILING(\$A\$1-8,7)+8

**Case 2 - If the Day falls on the same date, then next date** (Hence, in case of 2-Jan-17, next Monday would be 9-Jan-17 only)

<b>Next Mon</b>	=CEILING(\$A\$1-1,7)+2
<b>Next Tue</b>	=CEILING(\$A\$1-2,7)+3
<b>Next Wed</b>	=CEILING(\$A\$1-3,7)+4
<b>Next Thu</b>	=CEILING(\$A\$1-4,7)+5
<b>Next Fri</b>	=CEILING(\$A\$1-5,7)+6
<b>Next Sat</b>	=CEILING(\$A\$1-6,7)+7
<b>Next Sun</b>	=CEILING(\$A\$1-7,7)+8

## 47. Find the Previous Week of the Day

There are 2 scenarios in this case. For example, if today's date is 2-Jan-17 (Monday) and I try to find the previous Monday, I can get either 2-Jan-17 or 26-Dec-16 as per need. For Tuesday to Sunday, it is not a problem as they come prior to 2-Jan-17 only.

**Case 1 - If the Day falls on the same date, then that very date** (Hence, in case of 2-Jan-17, previous Monday would be 2-Jan-17 only)

<b>Previous Mon</b>	=CEILING(\$A\$1-8,7)+2
<b>Previous Tue</b>	=CEILING(\$A\$1-9,7)+3
<b>Previous Wed</b>	=CEILING(\$A\$1-10,7)+4
<b>Previous Thu</b>	=CEILING(\$A\$1-11,7)+5
<b>Previous Fri</b>	=CEILING(\$A\$1-12,7)+6
<b>Previous Sat</b>	=CEILING(\$A\$1-13,7)+7
<b>Previous Sun</b>	=CEILING(\$A\$1-14,7)+8

**Case 2 - If the Day falls on the same date, then previous date** (Hence, in case of 2-Jan-17, previous Monday would be 26-Dec-16 only)

<b>Previous Mon</b>	=CEILING(\$A\$1-9,7)+2
<b>Previous Tue</b>	=CEILING(\$A\$1-10,7)+3
<b>Previous Wed</b>	=CEILING(\$A\$1-11,7)+4
<b>Previous Thu</b>	=CEILING(\$A\$1-12,7)+5
<b>Previous Fri</b>	=CEILING(\$A\$1-13,7)+6
<b>Previous Sat</b>	=CEILING(\$A\$1-14,7)+7
<b>Previous Sun</b>	=CEILING(\$A\$1-15,7)+8



## 48. Count Cells Starting (or Ending) with a particular String

1. Say you want to count all cells starting with C

```
=COUNTIF(A1:A10,"c*")
```

c\* is case insensitive. Hence, it will count cells starting with both c or C.

Suppose you want to find all cells starting with Excel.

```
=COUNTIF(A1:A10,"excel*")
```

2. For ending

```
=COUNTIF(A1:A10,"*c")
```

c\* is case insensitive. Hence, it will count cells starting with both c or C.

Suppose you want to find all cells starting with Excel.

```
=COUNTIF(A1:A10,"*excel")
```

## 49. Count No. of Cells Having Numbers Only

COUNT function counts only those cells which are having numbers.

Assuming your range is A1:A10, use following formula

```
=COUNT(A1:A10)
```

## 50. Reverse a String

Suppose cell A1:="qwerty" and you want to reverse it

```
=TEXTJOIN(,,MID(A1,LEN(A1)-SEQUENCE(LEN(A1))+1,1))
```

```
=TEXTJOIN(,,MID(A1,LEN(A1)-ROW(INDIRECT("1:"&LEN(A1)))+1,1))
```

## 51. Reverse Number String

Suppose cell A1 has number 16709 and you need to reverse it i.e. you want 90761, then following formula can be used

```
=SUMPRODUCT(MID(A1,SEQUENCE(LEN(A1)),1)*10^SEQUENCE(LEN(A1))/10)
```

```
=--TEXTJOIN(,,MID(A1,LEN(A1)-SEQUENCE(LEN(A1))+1,1))
```

```
=--TEXTJOIN(,,MID(A1,LEN(A1)-ROW(INDIRECT("1:"&LEN(A1)))+1,1))
```

## 52. Reverse a String of Words

Suppose you have following string of words in cell A2 - "Moscow, London, Paris, Delhi, Washington, Miami, Detroit, Berlin"

You want to reverse this string of words and want following output.

"Berlin, Detroit, Miami, Washington, Delhi, Paris, London, Moscow"

FILTERXML will come to your rescue. You can use following formula for this

```
=TEXTJOIN(" ",,SORTBY(FILTERXML("<t><s>"&SUBSTITUTE(A2,"","</s><s>")&"</s></t>","//s"),SEQUENCE(LEN(A2)-LEN(SUBSTITUTE(A2,"","")+1),-1))
```

```
=TEXTJOIN(" ",,INDEX(FILTERXML("<t><s>"&SUBSTITUTE(A2,"","</s><s>")&"</s></t>","//s"),LEN(A2)-LEN(SUBSTITUTE(A2,"",""))+2-ROW(INDIRECT("1:"&LEN(A2)-LEN(SUBSTITUTE(A2,"","")+1))))
```

## 53. Reverse a Range of Cells

Suppose you have following words in range A2:A11 and you want to reverse the sequence

	A	B	C	D
1	<b>Names</b>			<b>Names</b>
2	James			Charles
3	Robert			Thomas
4	John			Joseph
5	Michael			Richard
6	William			David
7	David			William
8	Richard			Michael
9	Joseph			John
10	Thomas			Robert
11	Charles			James

Following formula can be used in D2 which will spill into cells down below

```
=INDEX(A2:A11,COUNTA(A2:A11)-SEQUENCE(COUNTA(A2:A11))+1)
```

```
=INDEX(A2:A11,COUNTA(A2:A11)-ROW(INDIRECT("1:"&COUNTA(A2:A11)))+1)
```

```
=SORTBY(FILTER(A2:A11,A2:A11<>""),SEQUENCE(COUNTA(A2:A11)),-1)
```

## 54. Get First Row Number in a Range

Suppose your range given is D15:Q99 and you want to extract the first row number which is 15 here, then use following formula

```
=@ROW(D15:Q99)
```

=MIN(ROW(D15:Q99))

## 55. Get Last Row Number in a Range

Suppose your range given is D15:Q99 and you want to extract the last row number which is 99 here, then use following formula

=@SORT(ROW(D15:Q99),,-1)

=MAX(ROW(D15:Q99))

## 56. Get First Column Number in a Range

Suppose your range given is D15:Q99 and you want to extract the first column number which is 4 (4 means D) here, then use following formula

=@COLUMN(D15:Q99)

=MIN(COLUMN(D15:Q99))

## 57. Get Last Column Number in a Range

Suppose your range given is D15:Q99 and you want to extract the last column number which is 17 (17 means Q) here, then use following formula

=MAX(COLUMN(D15:Q99))

## 58. Extract Unique Characters

Suppose, cell A1 has the value "abracadabra" and you want to extract unique characters from this in i.e. the answer should be "abrcd". Then following formula can be used (provided you have Dynamic Array functionality)

=TEXTJOIN(,UNIQUE(MID(A1,SEQUENCE(LEN(A1)),1)))

=TEXTJOIN(,UNIQUE(MID(A1,ROW(INDIRECT("1:"&LEN(A1))),1)))

## 59. Test whether a range contains only numbers

Use below formula to test whether a given range say A1:A10 contains only numbers

=ISNUMBER(--TEXTJOIN("",1,A1:A10))

## 60. Test whether a range contains only English alphabets

Use below formula to test whether a given range say A1:A10 contains only English alphabets

```
=NOT(ISERROR(SUM(MATCH(UPPER(A1:A10),CHAR(ROW(65:90)),0))))
```

If you want to ignore blanks

```
=NOT(ISERROR(SUM(MATCH(UPPER(FILTER(A1:A10,A1:A10<>"")),CHAR(ROW(65:90)),0))))
```

## 61. Count No. of Cells which are containing only Characters

Hence, if your cell is having a number 2.23, it will not be counted as it is a number.

Use below formula considering your range is A1:A10

```
=COUNTIF(A1:A10,"*")
```

## 62. Number of Characters in a String without considering blanks

Say, you have a string like Vijay A. Verma and I need to know how many characters it has. In this case, it has 12 including decimal and leaving blanks aside.

Use below formula for the same -

```
=LEN(SUBSTITUTE(A1," ",""))
```

## 63. Number of times a character appears in a string

Suppose you want to count the number of times, character "a" appears in a string

```
=LEN(A1)-LEN(SUBSTITUTE(LOWER(A1),"a",""))
```

## 64. Count Non Numbers in a String

Suppose you have a string "abc123def45cd" and you want to count non numbers in this.

If your string is in A1, use following formula in A1

```
=COUNT(IF(ISNUMBER(--MID(A1,SEQUENCE(LEN(A1)),1)), " ",0))-(A1="")
```

```
=COUNT(IF(ISNUMBER(--MID(A1,ROW(INDIRECT("1:"&LEN(A1))),1)), " ",0))-(A1="")
```

```
=SUMPRODUCT(--NOT(ISNUMBER(--MID(A1,SEQUENCE(LEN(A1)),1))))-(A1="")
```

```
=SUMPRODUCT(--NOT(ISNUMBER(--MID(A1,ROW(INDIRECT("1:"&LEN(A1))),1))))-(A1="")
```

## 65. Count Numbers in a String

Suppose you have a string "abc123def43cd" and you want to count numbers in this.

If your string is in A1, use following formula –

```
=COUNT(--MID(A1,SEQUENCE(LEN(A1)),1))
```

```
=COUNT(--MID(A1,ROW(INDIRECT("1:"&LEN(A1))),1))
```

```
=SUMPRODUCT(--ISNUMBER(--MID(A1,SEQUENCE(LEN(A1)),1)))
```

```
=SUMPRODUCT(--ISNUMBER(--MID(A1,ROW(INDIRECT("1:"&LEN(A1))),1)))
```

```
=SUMPRODUCT(LEN(A1)-LEN(SUBSTITUTE(A1,ROW(1:10)-1,"")))
```

## 66. Count only Alphabets in a String

Suppose you have a string "Ab?gh123def%h\*" and you want to count only Alphabets.

Suppose your string is in A1, put following formula for this.

```
=SUMPRODUCT(LEN(A1)-  
LEN(SUBSTITUTE(UPPER(A1),CHAR(ROW(INDIRECT("65:90"))),"")))
```

```
=SUMPRODUCT(--(ABS(77.5-  
CODE(MID(UPPER(A1),ROW(INDIRECT("A1:A"&LEN(A1))),1)))<13))
```

## 67. Generate English Alphabets

To generate English alphabets A...Z

```
=CHAR(SEQUENCE(26,,65))
```

```
=CHAR(ROW(65:90))
```

To generate English alphabets a...z

```
=CHAR(SEQUENCE(26,,97))
```

```
=CHAR(ROW(97:122))
```

Above formulas will generate English alphabets in a column. To generate in a row

```
=CHAR(SEQUENCE(,26,65))
```

```
=CHAR(SEQUENCE(,26,97))
```

```
=TRANSPOSE(CHAR(ROW(65:90)))
```

=TRANSPOSE(CHAR(ROW(97:122)))

## 68. Convert English Alphabets to Numbers

There may be scenarios where you need to convert alphabets a, b to y, z to 1, 2 to 25, 26 (Or A, B to Y, Z to 1, 2 to 25, 26) You can use one of the following formulas to do it.

=CODE(LOWER(A1))-96

=CODE(UPPER(A1))-64

## 69. Convert Numbers to English Alphabets

There may be scenarios where you need to convert numbers 1, 2 to 25, 26 to a, b to y, z (Or to A, B to Y, Z) You can use one of the following formulas to do it

(first formula is for lower case and second one is for upper case)

=CHAR(A1+96)

=CHAR(A1+64)

## 70. Extract nth Word from Front

Suppose you have following string – "Moscow, London, Paris, Delhi, Washington, Miami, Detroit, Berlin"

And you want to retrieve nth word from front.

You can use following FILTERXML formula for this

=FILTERXML("<t><s>"&SUBSTITUTE(A2," ","</s><s>")&"</s></t>","//s[N]")

Where N need to be replaced with the word number which you want.

Hence, if you wanted 2nd word, then replace that with 2

=FILTERXML("<t><s>"&SUBSTITUTE(A2," ","</s><s>")&"</s></t>","//s[2]")

Note – In my string, comma followed by a space is separator. If your separator is different say space, replace red part in SUBSTITUTE(A2," ","</s><s>"). Hence, if your separator is simply a space, then

=FILTERXML("<t><s>"&SUBSTITUTE(A2," ","</s><s>")&"</s></t>","//s[N]")

## 71. Extract nth Word from Back

Suppose you have following string

Moscow, London, Paris, Delhi, Washington, Miami, Detroit, Berlin

And you want to retrieve nth word from back.

You can use following FILTERXML formula for this

Now you want to retrieve nth word from back. You can use following FILTERXML formula for this

```
=FILTERXML("<t><s>"&SUBSTITUTE(A2,"","</s><s>")&"</s></t>","//s[("&LEN(A2)-LEN(SUBSTITUTE(A2,""," "))+2-N&"]")
```

Hence, if you want last word, replace N with 1. If you need 2nd last word, replace N with 2. Hence for 3rd last word, the formula would be (after replacing N with 3)

```
=FILTERXML("<t><s>"&SUBSTITUTE(A2,"","</s><s>")&"</s></t>","//s[("&LEN(A2)-LEN(SUBSTITUTE(A2,""," "))+2-3&"]")
```

Note – Here again, separator in SUBSTITUTE need to be replaced appropriately if you are not using my separator in your string.

## 72. Extract first two words OR first n words

You can use following formula to extract first 2 words from a text string

```
=IFERROR(LEFT(A2,FIND(" ",A2&" ",FIND(" ",A2)+1)-1), "")
```

A generic formula to extract first 2 words which can be extended to n words

```
=IFERROR(LEFT(A2,FIND(REPT(" ",LEN(A2)),SUBSTITUTE(A2&" "," ",REPT(" ",LEN(A2)),2))-1), "")
```

```
=IFERROR(REPLACE(A2,FIND(REPT(" ",LEN(A2)),SUBSTITUTE(A2&" "," ",REPT(" ",LEN(A2)),2)),LEN(A2),""), "")
```

To extract first n words, replace 2 above which is highlighted in bold. Hence, if you want to extract first 3 words

```
=IFERROR(LEFT(A2,FIND(REPT(" ",LEN(A2)),SUBSTITUTE(A2&" "," ",REPT(" ",LEN(A2)),3))-1), "")
```

```
=IFERROR(REPLACE(A2,FIND(REPT(" ",LEN(A2)),SUBSTITUTE(A2&" "," ",REPT(" ",LEN(A2)),3)),LEN(A2),""), "")
```

## 73. Extract last two words OR last n words

To extract last 2 words, use below formula

```
=IF(ISNUMBER(FIND(" ",A2)),TRIM(RIGHT(SUBSTITUTE(A2," ",REPT(" ",LEN(A2))),2*LEN(A2))), "")
```

To make a generic formula, to extract last 2 words

```
=IF(COUNTIF(A2,"* *")>0,TRIM(RIGHT(SUBSTITUTE(A2," ",REPT(" ",LEN(A2))),2*LEN(A2))),"
```

Now, you would need to change \* \* and 2 only which are marked in Red. If you need to replace last 3 words, then \* \* would be replaced with \* \* \* and 2 with 3. Hence, formula would become

```
=IF(COUNTIF(A2,"* * *")>0,TRIM(RIGHT(SUBSTITUTE(A2," ",REPT(" ",LEN(A2))),3*LEN(A2))),"
```

Hence, if you want to extract last word only

```
=IF(COUNTIF(A2,"*")>0,TRIM(RIGHT(SUBSTITUTE(A2," ",REPT(" ",LEN(A2))),1*LEN(A2))),"
```

Note, actually for last word, simply use the second component of the formula (just to make it simpler)

```
=TRIM(RIGHT(SUBSTITUTE(A2," ",REPT(" ",LEN(A2))),1*LEN(A2)))
```

If you are not a fan of \* \* technique and want to replace numbers only, use below formula where 2 can be replaced with any number (2 is for extracting last 2 words)

```
=IF((LEN(A2)-LEN(SUBSTITUTE(A2," ","")))>=2-1,TRIM(RIGHT(SUBSTITUTE(A2," ",REPT(" ",LEN(A2))),2*LEN(A2))),"
```

## 74. Most Frequently Occurring Value in a Range

Assuming, your range is A1:A10, enter the below formula

```
=INDEX(A1:A10,MODE(MATCH(A1:A10,A1:A10,0)))
```

Note – If range A1:A10 contains only numbers, then following formula is enough

```
=MODE(A1:A10)
```

## 75. COUNTIF with OR Condition

	A	B
1	Zone	Subzone
2	North	A
3	North	B
4	North	B
5	East	A
6	West	A
7	North	C
8	South	B
9	North	C
10	North	B

For the given range, say you want to count Subzones A and B for North Zone, then you can use following formula

```
=SUM(COUNTIFS(A2:A10,"North",B2:B10,{"A","B"}))
```



## 76. SUMIF with OR Condition

	A	B	C
1	<b>Class</b>	<b>Subject</b>	<b>Marks</b>
2		1 English	64
3		1 English	31
4		1 French	13
5		1 Latin	38
6		1 English	78
7		2 French	13
8		2 English	89
9		2 English	19
10		2 French	64

For the given range, say you want to SUM the Marks for class 1 and Subjects = English and French. Then you can use following formula

=SUM(SUMIFS(C2:C10,A2:A10,1,B2:B10,{"English","French"}))

## 77. COUNTIF on Filtered List

You can use SUBTOTAL to perform COUNT on a filtered list but COUNTIF can not be done on a filtered list. Below formula can be used to perform COUNTIF on a filtered list

=SUMPRODUCT(SUBTOTAL(3,OFFSET(B2,ROW(B2:B20)-ROW(B2),))\*(B2:B20>14))

Here B2:B20>14 is like a criterion in COUNTIF (=COUNTIF(B2:B20,">14"))

## 78. SUMIF on Filtered List

You can use SUBTOTAL to perform SUM on a filtered list but SUMIF can not be done on a filtered list. Below formula can be used to perform SUMIF on a filtered list

=SUMPRODUCT(SUBTOTAL(9,OFFSET(B2,ROW(B2:B20)-ROW(B2),))\*(B2:B20>14))

Here B2:B20>14 is like a criterion in SUMIF.

## 79. Extract First Name from Full Name

=LEFT(A1,FIND(" ",A1&" ")-1)

## 80. Extract Last Name from Full Name

=TRIM(RIGHT(SUBSTITUTE(A1," ",REPT(" ",LEN(A1))),LEN(A1)))

## 81. Extract the Initial of Middle Name

Suppose, you have a name John Doe Smith and you want to show D as middle initial. Assuming, your data is in A1, you may use following formula

=IF(COUNTIF(A1,"\* \* \*"),MID(A1,FIND(" ",A1)+1,1),"")

If name is of 2 or 1 words, the result will be blank. This works on 3 words name only as middle can be decided only for 3 words name.

## 82. Extract Middle Name from Full Name

```
=IF(COUNTIF(A1,"* * *"),MID(A1,FIND(" ",A1)+1,FIND(" ",A1,FIND(" ",A1)+1)-(FIND(" ",A1)+1)), "")
```

```
=IF(COUNTIF(A1,"* * *"),TRIM(MID(SUBSTITUTE(A1," ",REPT(" ",LEN(A1))),2),FIND(" ",A1)+1,LEN(A1))), "")
```

```
=IF(COUNTIF(A1,"* * *"),LEFT(REPLACE(A1,1,FIND(" ",A1),""),FIND(" ",REPLACE(A1,1,FIND(" ",A1),""))-1))
```

## 83. Remove Middle Name in Full Name

```
=IF(COUNTIF(A1,"* * *"),LEFT(A1,FIND(" ",A1&" "))&TRIM(RIGHT(SUBSTITUTE(A1," ",REPT(" ",LEN(A1))),LEN(A1))), "")
```

```
=IF(COUNTIF(A1,"* * *"),REPLACE(A1,FIND(" ",A1)+1,FIND(" ",A1,FIND(" ",A1)+1)-FIND(" ",A1),""))
```

## 84. Extract Integer and Decimal Portion of a Number

To extract Integer portion, one of the below can be used -

```
=INT(A1)
```

```
=TRUNC(A1)
```

Positive value in A1 - If A1 contains 84.65, then answer would be 84 in both cases

**Negative value in A1 - If A1 contains -84.65, then answer would be -85 in case of INT and -84 in case of TRUNC.**

If you want only +ve value whether value in A1 is -ve or +ve, the formula can have many variants. Notice answers for negative values in case INT.

	84.65	-84.65
INT(ABS(A1))	84	84
TRUNC(ABS(A1))	84	84
ABS(INT(A1))	84	85
ABS(TRUNC(A1))	84	84

To extract Decimal portion -

```
=MOD(ABS(A1),1)
```

=ABS(A1)-INT(ABS(A1))

Positive value in A1 - If A1 contains 84.65, then answer would be 0.65.

Negative value in A1 - If A1 contains -24.39, then answer would be 0.39.

## 85. Maximum Times a Particular Entry Appears Consecutively

	A	B
1	<b>Region</b>	<b>Sales Man</b>
2	A	a
3	A	a
4	g	b
5	g	b
6	g	b
7	g	c
8	A	c
9	A	a
10	B	m
11	B	n
12	A	o
13	A	p
14	A	d
15	B	d
16	B	e
17	B	e

Suppose, we want to count maximum times "A" appears consecutively,

=MAX(FREQUENCY(IF(A2:A20="A",ROW(A2:A20)),  
IF(A2:A20<>"A",ROW(A2:A20))))

## 86. Get File Name through Formula

Before getting this, make sure that you file has been saved at least once as this formula is dependent upon the file path name which can be pulled out by CELL function only if file has been saved at least once.

=CELL("filename",\$A\$1)

## 87. Get Workbook Name through Formula

Before getting this, make sure that you file has been saved at least once as this formula is dependent upon the file path name which can be pulled out by CELL function only if file has been saved at least once.

=REPLACE(LEFT(CELL("filename",\$A\$1),FIND("]",CELL("filename",\$A\$1))-1),1,FIND("[",CELL("filename",\$A\$1)),"")

## 88. Get Sheet Name through Formula

Before getting this, make sure that you file has been saved at least once as this formula is dependent upon the file path name which can be pulled out by CELL function only if file has been saved at least once.

Use following formula -

```
=REPLACE(CELL("filename",A1),1,FIND("]",CELL("filename",A1)),"")
```

Make sure that A1 is used in the formula. If it is not used, it will extract sheet name for the last active sheet which may not be one which we want.

If you want the sheet name for last active sheet only, then formula would become

```
=REPLACE(CELL("filename"),1,FIND("]",CELL("filename")), "")
```

## 89. Get Workbook's Directory from Formula

Before getting this, make sure that you file has been saved at least once as this formula is dependent upon the file path name which can be pulled out by CELL function only if file has been saved at least once.

If your workbook is located in say C:\Excel\MyDocs, the formula to retrieve the directory for this would be

```
=LEFT(CELL("filename",A1),FIND("[",CELL("filename",A1))-2)
```

## 90. Perform Multi Column VLOOKUP

You know VLOOKUP, one of the most loved function of Excel. The syntax is VLOOKUP(lookup\_value,table\_array,col\_index\_num,range\_lookup)

Here lookup\_value can be a single value not multiple values.

Now, you are having a situation where you want to do vlookup with more than 1 values. For the purpose of illustrating the concept, let's say we have 2 values to be looked up.

Below is your lookup table and you want to look up for Emp - H and Gender - F for Age.

	A	B	C
1	Emp	Gender	Age
2	A	M	37
3	A	F	35
4	C	F	39
5	D	F	40
6	E	M	37
7	E	M	39
8	G	M	34
9	H	F	25
10	H	M	35
11	K	F	35
12	K	M	38

```
=XLOOKUP(1,(A2:A12=F2)*(B2:B12=G2),C2:C12)
```

```
=INDEX(FILTER(A2:C12,(A2:A12=F2)*(B2:B12=G2)),3)
```

```
=INDEX(C2:C12,MATCH(1,--((A2:A12=F2)*(B2:B12=G2)*(ROW(A2:A12)-ROW(A2)+1)<>0),0))
```

## Concatenation Approach

=XLOOKUP(F2&"@@@"&G2,A2:A12&"@@@"&B2:B12,C2:C12,"")

=INDEX(C2:C12,MATCH(F2&"@@@"&G2,A2:A12&"@@@"&B2:B12,0))

@@@ can be replaced by any characters which should not be part of those columns.

By concatenation, you can have as many columns as possible.

**CAUTION** - Result of entire concatenation should not be having length more than 255. Hence, F2&"@@@"&G2 should not have more than 255 characters.

Another alternative is to use below formula -

=INDEX(C2:C12,MATCH(1,--NOT(ISLOGICAL(IF(A2:A12=F2,IF(B2:B12=G2,C2:C12))))),0))

**Note** - To handle Not Found condition, you can enclose all above formulas in an IFERROR block.

## 91. VLOOKUP from Right to Left

(Made redundant by XLOOKUP but useful for older version)

VLOOKUP always looks up from Left to Right. Hence, in the below table, I can find Date of Birth of Naomi by giving following formula -

=VLOOKUP("Naomi",B:D,3,0)

	A	B	C	D	E	F	G	H	I
1	Emp ID	First Name	Gender	Date of Birth	Date of Join	Salary	State	Zip	Region
2	250945	Terisa	F	5/23/1984	12/11/2009	64667	VA	22651	South
3	778777	Milton	M	10/10/1986	1/6/2014	46688	TX	78344	South
4	962153	Lonnie	F	1/13/1972	7/23/2009	78661	MN	56088	Midwest
5	833630	Naomi	F	11/25/1961	2/27/1990	118632	AR	71855	South
6	565499	Cathryn	F	8/13/1979	11/30/2009	181053	WA	99349	West
7	608769	Sylvester	M	5/24/1969	6/28/1999	76790	LA	71418	South
8	262325	Jamaal	M	3/16/1968	8/30/1999	170583	IN	46866	Midwest
9	562671	Leslie	F	8/8/1975	7/14/2007	136188	MN	56178	Midwest
10	711925	Gaynell	F	7/30/1990	1/20/2013	88425	MD	20724	South
11	613284	Loria	F	7/29/1992	5/20/2015	194557	OH	44282	Midwest
12	917399	Florance	F	3/31/1965	7/18/1986	177637	CA	94307	West
13	369366	Mathew	M	6/11/1993	8/16/2015	193972	MS	39113	South
14	642751	Scottie	M	12/24/1966	9/3/2015	198240	MO	63876	Midwest
15	323627	Timmy	M	5/19/1993	9/21/2014	175621	NM	87413	West
16	833463	Warren	M	5/13/1981	8/30/2011	142919	CA	95667	West
17	146564	Delta	F	6/19/1984	10/10/2011	132695	WV	25685	South
18	290540	Hal	M	12/5/1980	3/21/2009	118357	IA	51201	Midwest
19	219886	Alberto	M	3/16/1958	3/6/1992	126401	VA	23054	South

But, If I have to find Emp ID corresponding to Naomi, I can not do it through VLOOKUP formula. To perform VLOOKUP from Right to Left, you will have to use INDEX / MATCH combination. Hence, you will have to use following formula –

```
=INDEX(A:A,MATCH("Naomi",B:B,0))
```

If you use XLOOKUP, the formula would simply be

```
=XLOOKUP("Naomi",B:B,A:A)
```

## 92. Case Sensitive VLOOKUP

Suppose you have data like below table and you want to do a case sensitive VLOOKUP

	A	B
1	<b>Name</b>	<b>Salary</b>
2	Lisa	3500
3	Mira	2300
4	Sara	4300
5	SARA	3200
6	Sarah	4500

If perform a regular VLOOKUP on SARA, I would get the answer 4300. But in a case sensitive VLOOKUP, answer should be 3200. You may use below formula for Case Sensitive VLOOKUP

```
=XLOOKUP(1,--EXACT("SARA",A2:A10),B2:B10)
```

```
=INDEX(FILTER(A2:B6,EXACT(A2:A6,"SARA")),2)
```

```
=INDEX(B2:B10,MATCH(TRUE,EXACT("SARA",A2:A10),0))
```

## 93. Rank within the Groups

Suppose your have data like below table and you want to know rank of students.

Student	School	Marks	Rank
Student 1	School 1	92	
Student 2	School 2	64	
Student 3	School 3	73	
Student 4	School 4	72	
Student 5	School 5	46	
Student 6	School 6	70	
Student 7	School 7	55	
Student 8	School 8	21	
Student 9	School 9	91	
Student 10	School 10	27	
Student 11	School 1	91	
Student 12	School 2	84	
Student 13	School 3	40	
Student 14	School 4	70	
Student 15	School 5	51	
Student 16	School 6	37	
Student 17	School 7	53	
Student 18	School 8	54	
Student 19	School 9	53	
Store_20	School 10	59	

You will simple put following formula in D2

```
=RANK(C2,C2:C100)
```

But what if you are asked to produce rank of students within each school. Hence, every school's rank will start with 1..

Put following formula in D2 for that case for Descending order ranking. (For ascending order, replace ">" with "<" without quote marks)

```
=SUMPRODUCT((B$2:B$100=B2)*(C$2:C$100>C2))+1
```



To remove numbers from a string (for example Vij1aY A. V4er7ma8 contains numbers which are not required), use below formula assuming string is in A1 cell –

```
=TEXTJOIN("",IF(ISERROR(--MID(A1,SEQUENCE(LEN(A1)),1)),MID(A1,SEQUENCE(LEN(A1)),1),""))
```

```
=TEXTJOIN("",IF(ISERROR(--MID(A1,ROW(INDIRECT("A1:A"&LEN(A1))),1)),MID(A1,ROW(INDIRECT("A1:A"&LEN(A1))),1),""))
```

```
=SUBSTITUTE(SUBSTITUTE(SUBSTITUTE(SUBSTITUTE(SUBSTITUTE(SUBSTITUTE(SUBSTITUTE(SUBSTITUTE(SUBSTITUTE(SUBSTITUTE(A1,1,""),2,""),3,""),4,""),5,""),6,""),7,""),8,""),9,""),0,"")
```

Note - Since this formula is in multiple lines, hence you will have to copy this in Formula Bar. If you copy this formula in a cell, it will copy this in three rows.

## 98. Roman Representation of Numbers

Use ROMAN function.

Hence ROMAN(56) will give LVI.

ROMAN works only for numbers 1 to 3999.

## 99. Sum Bottom N Values in a Range

Suppose you have numbers in range A1:A100 and you want to sum up bottom 10 values

```
=SUM(SMALL($A$1:$A$100,ROW(1:10)))
```

```
=SUMPRODUCT(SMALL($A$1:$A$100,ROW(1:10)))
```

```
=SUM(AGGREGATE(15,6,$A$1:$A$100,ROW(1:10)))
```

```
=SUMPRODUCT(AGGREGATE(15,6,$A$1:$A$100,ROW(1:10)))
```

In case, you want to ignore 0 values (and blanks)

```
=SUM(SMALL(IF($A$1:$A$100<>0,$A$1:$A$100),ROW(1:10)))
```

```
=SUMPRODUCT(SMALL(IF($A$1:$A$100<>0,$A$1:$A$100),ROW(1:10)))
```

Both the above formulas will function only if there are at least N values as per ROW(1:N). Hence, for above formulas, it would work only if there are at least 10 numbers in A1 to A100.

To overcome this limitation -



Enter the below formulas

```
=SUM(IFERROR(SMALL($A$1:$A$100,ROW(1:10)),0))
```

```
=SUM(IFERROR(SMALL(IF($A$1:$A$100<>0,$A$1:$A$100),ROW(1:10)),0))
```

```
=SUMPRODUCT(AGGREGATE(15,6,$A$1:$A$100/($A$1:$A$100<>0),ROW(1:10)))
```

```
=SUM(AGGREGATE(15,6,$A$1:$A$100/($A$1:$A$100<>0),ROW(1:10)))
```

## 100. Sum Every Nth Row

If your numbers are in range A1:A100, use below formula

```
=SUM((A1:A100)*(MOD(ROW(A1:A100)-ROW(A1)+1,2)=0))
```

```
=SUMPRODUCT((A1:A100)*(MOD(ROW(A1:A100)-ROW(A1)+1,2)=0))
```

Above formula is for every 2nd row. Replace 2 with N. Hence, for every 5th row –

```
=SUM((A1:A100)*(MOD(ROW(A1:A100)-ROW(A1)+1,5)=0))
```

```
=SUMPRODUCT((A1:A100)*(MOD(ROW(A1:A100)-ROW(A1)+1,5)=0))
```

This is a generic formula and will work for any range. If your range is B7:B50, your formula would become

```
=SUM((B7:B50)*(MOD(ROW(B7:B50)-ROW(B7)+1,2)=0))
```

```
=SUMPRODUCT((B7:B50)*(MOD(ROW(B7:B50)-ROW(B7)+1,2)=0))
```

## 101. Sum Every Even Row

If your range is A1:A100, use following formula.

```
=SUM((A1:A100)*(MOD(ROW(A1:A100)-ROW(A1)+1,2)=0))
```

```
=SUMPRODUCT((A1:A100)*(MOD(ROW(A1:A100)-ROW(A1)+1,2)=0))
```

This is a generic formula, hence if your range is B7:B50, your formula will become

```
=SUM((B7:B50)*(MOD(ROW(B7:B50)-ROW(B7)+1,2)=0))
```

```
=SUMPRODUCT((B7:B50)*(MOD(ROW(B7:B50)-ROW(B7)+1,2)=0))
```

## 102. Sum Every Odd Row

If your range is A1:A100, use following formula.

```
=SUM((A1:A100)*(MOD(ROW(A1:A100)-ROW(A1)+1,2)<>0))
```

=SUMPRODUCT((A1:A100)\*(MOD(ROW(A1:A100)-ROW(A1)+1,2)<>0))

This is a generic formula, hence if your range is B7:B50, your formula will become

=SUM((B7:B50)\*(MOD(ROW(B7:B50)-ROW(B7)+1,2)<>0))

=SUMPRODUCT((B7:B50)\*(MOD(ROW(B7:B50)-ROW(B7)+1,2)<>0))

### 103. Sum Top N values in a Range

Suppose you have numbers in range A1:A100 and you want to sum up top N values

=SUM(LARGE(\$A\$1:\$A\$100,ROW(1:10)))

=SUMPRODUCT(LARGE(\$A\$1:\$A\$100,ROW(1:10)))

=SUM(AGGREGATE(14,6,\$A\$1:\$A\$100,ROW(1:10)))

=SUMPRODUCT(AGGREGATE(14,6,\$A\$1:\$A\$100,ROW(1:10)))

In case, you want to ignore 0 values (and blanks)

=SUM(LARGE(IF(\$A\$1:\$A\$100<>0,\$A\$1:\$A\$100),ROW(1:10)))

=SUMPRODUCT(LARGE(IF(\$A\$1:\$A\$100<>0,\$A\$1:\$A\$100),ROW(1:10)))

Both the above formulas will function only if there are at least N values as per ROW(1:N). Hence, for above formulas, it would work only if there are at least 10 numbers in A1 to A100. To overcome this limitation –

=SUM(IFERROR(LARGE(\$A\$1:\$A\$100,ROW(1:10)),0))

=SUM(IFERROR(LARGE(IF(\$A\$1:\$A\$100<>0,\$A\$1:\$A\$100),ROW(1:10)),0))

=SUM(AGGREGATE(14,6,\$A\$1:\$A\$100/(\$A\$1:\$A\$100<>0),ROW(1:10)))

=SUMPRODUCT(AGGREGATE(14,6,\$A\$1:\$A\$100/(\$A\$1:\$A\$100<>0),ROW(1:10)))

### 104. We have AVERAGEIF. What about MEDIANIF and MODEIF?

Excel doesn't provide MEDIANIF and MODEIF. You will have to use Array formulas to achieve these functionality. Let's assume that our data is like below –

	A	B	C
1	<b>Gender</b>	<b>Income</b>	
2	M	40000	
3	F	70000	
4	M	10000	
5	F	70000	
6	M	50000	
7	F	40000	
8	M	10000	
9	F	30000	
10	M	80000	
11	F	80000	
12	M	60000	
13	F	50000	
14			
15			
16	Median	45000	{=MEDIAN(IF(A2:A13="M",B2:B13))}
17	Mode	10000	{=MODE(IF(A2:A13="M",B2:B13))}

To calculate MEDIANIF and MODEIF, enter below formulas

=MEDIAN(IF(A2:A13="M",B2:B13))

=MODE(IF(A2:A13="M",B2:B13))

#### For MEDIANIF

=AGGREGATE(16,6,(B1:B13)/(A1:A13="m"),50%)

#### For MODEIF

=INDEX(B1:B20,MATCH(MAX(INDEX((COUNTIF(B1:B20,B1:B20)\*(A1:A20="m")),,)),INDEX((COUNTIF(B1:B20,B1:B20)\*(A1:A20="m")),,)),0))

## 105. Calculate Geometric Mean by Ignoring 0 and Negative Values

Geometric Mean is a useful mean and is applied only for +ve values. Hence, you will need to ignore  $\leq 0$  values while calculating Geometric Mean. It is generally used where %ages are involved. For example, population growth for first year is 30%, for second year is 25% and for third year, it is 15%. Then Geometric Mean is used to calculate not Arithmetic Mean.

Generally, Geometric Mean is calculated by the formula =GEOMEAN(A1:A10)

It would give error if the range contains  $\leq 0$  values. There are various ways to deal with it and most commonly used way is to ignore  $\leq 0$  values while calculating Geometric Mean. To ignore  $\leq 0$  values, you must use an Array formula i.e. which must be entered by pressing CTRL+SHIFT+ENTER.

=GEOMEAN(IF(A1:A10>0,A1:A10))

The above formula takes into account only those values which are positive.

**Bonus Tip** - When %age growth are involved, you will need to use following ARRAY formula to calculate Geometric Mean -

```
=GEOMEAN(IF(A1:A10>0,(1+A1:A10)))-1
```

Don't forget to format your result as %age.

## 106. Generate GL Codes

To generate 20 numbers of GL codes like 1000, 1100, 1200 and so on., below formula can be used

```
=1000+(ROW(1:20)-1)*100
```

```
=SEQUENCE(20,,1000,100)
```

Above formula will generate GL codes in a column. To generate in a row

```
=SEQUENCE(,20,1000,100)
```

```
=TRANSPOSE(1000+(ROW(1:20)-1)*100)
```

## 107. Abbreviate Given Names

If you have names given like -

Smith Johnson

Liz lotte

Christy tiara Lewisk

John

And you need to produce abbreviations or acronyms for them like below in all capitals

Smith Johnson - SJ

Liz lotte - LT

Christy tiara Lewisk - CTL

john - J

Then you can use following formula for the same for upto 3 words in the name -

```
=UPPER(TRIM(LEFT(A1,1)&MID(A1,FIND(" ",A1&" ")+1,1)&MID(A1,FIND(" ",SUBSTITUTE(A1&" ", " ", "*"),2))+1,1)))
```

### Explanation for the formula

1. LEFT(A1,1) - Extracts the first letter from the first name

2. MID(A1,FIND(" ",A1&" ")+1,1) -

FIND(" ",A1&" ") - Find finds the first space in the given name to locate the start of the middle name. " " has been concatenated at the end of A1 so that if there is only first name, FIND will not give error as it will always find the blanks. +1 has been added to start the MID position from where the middle name starts.

3. MID(A1,FIND("\*",SUBSTITUTE(A1&" "," ","\*",2))+1,1))

SUBSTITUTE(A1&" "," ","\*",2) will replace the second blank with a \*, hence we can find the position of \* to locate the start of last name. As in 2 above, a double space " " has been added in A1 so that FIND always finds the second space. +1 has been added to start the MID position from where the last name starts.

4. TRIM will remove all blanks inserted because of 2 or 3.

5. UPPER will convert the string to all capitals.

Note - If you don't to use the concatenation of single space and double space as in 2 and 3, then IFERROR block can be used. In this case, the formula would become -

```
=UPPER(TRIM(LEFT(A1,1)&IFERROR(MID(A1,FIND("
",A1)+1,1),""))&IFERROR(MID(A1,FIND("*",SUBSTITUTE(A1," ","*",2))+1,1),"")))
```

**Note** - This technique can be used to extend up to many words. Only change will be in last block where you can replace 2 with 3, 4,5 and so on in IFERROR(MID(A1,FIND("\*",SUBSTITUTE(A1," ","\*",2))+1,1), "") for 4th, 5th, 6th words and concatenate them....Hence for upto 6 words, the formula would become

```
=UPPER(TRIM(LEFT(A1,1)&IFERROR(MID(A1,FIND("
",A1)+1,1),""))&IFERROR(MID(A1,FIND("*",SUBSTITUTE(A1," ","*",2))+1,1),""))
&IFERROR(MID(A1,FIND("*",SUBSTITUTE(A1,"
","*",3))+1,1),""))&IFERROR(MID(A1,FIND("*",SUBSTITUTE(A1," ","*",4))+1,1),""))
&IFERROR(MID(A1,FIND("*",SUBSTITUTE(A1," ","*",5))+1,1),"")))
```

## 108. Get Column Name for a Column Number

Let's suppose, you have a number in A1 and you want to get the column Name for that.

Hence, if A1=1, you want "A"

Hence, if A1 =26, you want "Z"

Hence, if A1=27, you want "AA" and so on.

The formula to derive the column name would be be -

```
=SUBSTITUTE(ADDRESS(1,A1,4),1,"")
```

## 109. Get Column Range for a Column Number

Let's suppose, you have a number in A1 and you want to get the column range for that.

Hence, if A1=1, you want "A:A"

Hence, if A1 =26, you want "Z:Z"

Hence, if A1=27, you want "AA:AA" and so on.

The formula to derive the column range would be be -

```
=SUBSTITUTE(ADDRESS(1,A1,4)&":"&ADDRESS(1,A1,4),1,"")
```

## 110. Find the nth Largest Number when there are duplicates

You know the LARGE function which can find the nth largest value. Hence, if you have a series like below -

10
12
18
9
7
14
22
24

And you give =LARGE(A1:A10,3), you get the answer as 18

Now, if we have a series like below

24
22
22
24
18
18
9
24

Now, you give =LARGE(A1:A10,3) and now the result is 24. The reason is that large function gives the nth largest value in a sorted array. Hence, LARGE function will sort the above array as {24,24,24,22,22,18,18,9} and 3rd largest is 24.

But actually you want the unique 3rd largest which is 18 as the answer.

The formula for such case would be

```
=LARGE(UNIQUE(A1:A10),3)
```

```
=LARGE(IF(FREQUENCY($A$1:$A$10,$A$1:$A$10)<>0,$A$1:$A$10),3)
```

## 111. COUNTIF for non-contiguous range

All of us love COUNTIF. And it is very easy to do - just say =COUNTIF("A1:A100", ">5") and it finds all the values within the range A1 to A100 which are greater than 5. But what if I wanted the result for only A3, A8 and it should omit other cells. Try putting in following formula -

```
=COUNTIF((A3, A8), ">5")
```

and it will give you #VALUE error.

A possible solution is

```
=(A3>5)+(A8>5)
```

What happens if you need to do for A3, A4, A5, A8, A24, A40, A45, A89. Now, you will have to use a formula like -

```
=(A3>5)+(A4>5)+(A5>5)+(A8>5)+(A24>5)+(A40>5)+(A45>5)+(A89>5)
```

The formula becomes cumbersome as the number of cells increase. In this case, you can use below formula. This single formula can take care of contiguous (like A3:A5) and non-contiguous ranges both -

```
=SUM(COUNTIF(INDIRECT({"A3:A5","A8","A24","A40","A45","A89"}),">5"))
```

## 112. Count the Number of Words in a Cell / Range

Suppose you have been given the following and you need to count the number of words in a cell or in a range.

Formula for calculating number of words in a cell -

```
=LEN(TRIM(A1))-LEN(SUBSTITUTE(TRIM(A1)," ",""))+(TRIM(A1)<>"")
```

Formula for calculating number of words in a range -

```
=SUMPRODUCT(LEN(TRIM(A1:A100))-LEN(SUBSTITUTE(TRIM(A1:A100)," ",""))+(TRIM(A1:A100)<>""))
```

	A	B	C	D	E	F
1	Smith	1	6			
2	John Smith	2				
3	John Vogt Smith	3				
4						
5						
6						
7						
8						
9						
10						

The table illustrates the number of words in a cell (column B) and the number of words in a range (column C). A blue box labeled "Number of Words in a Cell" points to cell B2 (2). A green box labeled "Number of Words in a Range" points to cell C2 (6).

## 113. Numerology Sum of the Digits aka Sum the Digits till the result is a single digit

In Numerology, it is often a task to add the digits till the result is a single digit. For example,

$$74 = 7 + 4 = 11 = 1 + 1 = 2$$

$$23 = 2 + 3 = 5$$

$$78 = 7 + 8 = 15 = 1 + 5 = 6$$

$$1234567 = 1 + 2 + 3 + 4 + 5 + 6 + 7 = 28 = 2 + 8 = 10 = 1 + 0 = 1$$

The formula to achieve the same is

$$=MOD(A1-1,9)+1$$

## 114. Generate Sequential Numbers and Repeat them

Suppose, you have been given the task to generate a sequence of numbers and repeat them.

For example -

1,2,3,4,1,2,3,4,1,2,3,4

You can use the below formula and drag down -

$$=MOD(ROWS(\$1:1)-1,4)+1$$

Replace 4 with with any other number to generate any other sequence. Hence, if you want to generate 1,2,3,4,5,6,7,8,9,10,1,2,3,4,5,6,7,8,9,10,1,2,3,4,5,6,7,8,9,10 then formula becomes -

$$=MOD(ROWS(\$1:1)-1,10)+1$$

The structure of the formula is

$$=MOD(ROWS(\$1:1)-1,X)+Y$$

X - Number of numbers

Y - Starting Number

Utilizing above formula, you want to generate the sequence 5,6,7,8,9,10,5,6,7,8,9,10,5,6,7,8,9,10, then use below formula (You need 6 numbers and starting number is 5)

$$=MOD(ROWS(\$1:1)-1,6)+5$$

## 115. Repeat a Number and Increment and Repeat....

Suppose, you have been given the task of repeating a number and increment that number and repeat it. For example -

1,1,1,1,2,2,2,2,3,3,3,3.....(Here, we are repeating it 4 times and incrementing and repeating 4 times again and so on)

Then you can use following formula

$$=ROUNDUP(ROWS(\$1:1)/4,0)$$

Suppose, you want to start the number with 5 not 1, then you can use below formula -



=ROUNDUP(ROWS(\$1:1)/4,0)+4

Hence, general structure of the formula is

=ROUNDUP(ROWS(\$1:1)/X,0)+Y-1

X - Number of times a particular number is repeated

Y - Starting Numbers

Hence, if you want to start with number 7 and you want to repeat it 5 times, then following formula should be used

=ROUNDUP(ROWS(\$1:1)/5,0)+6

## 116. Generate Non Repeating Random Numbers through Formula

Suppose, you want to generate non-repeating random numbers between 1 to 30, you can use following formula in A2 and drag down

=INDEX(UNIQUE(RANDARRAY(30-1+1,,1, 30, TRUE)), SEQUENCE(10))

In above formula, SEQUENCE(10) means that we want to generate 10 random numbers.

30-1+1 = Max - Min + 1 => These many numbers will be generated

1- Min and 30 - Max

TRUE - Integer only

=IFERROR(AGGREGATE(14,6,ROW(\$1:\$30)\*NOT(COUNTIF(\$A\$1:\$A1, ROW(\$1:\$30))),  
RANDBETWEEN(1,30-ROWS(\$1:1)+1)), "")

Note: \$A\$1:\$A1 is with reference to A2 as you put formula in A2 and dragged down. Suppose, you had put the formula in G4, this should be replaced with \$G\$3:\$G3.

If your starting and ending numbers are in B1 and C1, use below formula

=IFERROR(AGGREGATE(14,6,ROW(INDIRECT(\$B\$1&":"&\$C\$1))\*  
NOT(COUNTIF(\$A\$1:\$A1,ROW(INDIRECT(\$B\$1&":"&\$C\$1))))),  
RANDBETWEEN(\$B\$1,\$C\$1-ROWS(\$1:1)+1)), "")

For versions, prior to 2010 following basic construct can be used (Build error handling depending upon the version. For example, Excel 2007 will support IFERROR whereas 2003 supports ISERROR) -

=LARGE(INDEX(ROW(\$1:\$30)\*NOT(COUNTIF(\$A\$1:\$A1, ROW(\$1:\$30))),,,  
RANDBETWEEN(1,30-ROW(A1)+1))

## 117. Extract User Name from an E Mail ID

Assuming A1 has a mail ID say A1:=v.a.verma@gmail.com and you need to retrieve v.a.verma which is user name in the mail ID. Use following formula -

```
=IFERROR(LEFT(A1,SEARCH("@",A1)-1),"")
```

## 118. Extract Domain Name from an E Mail ID

If you want to retrieve domain name which in above example is gmail.com, use following formula –

```
=REPLACE(A1,1,SEARCH("@",A1)+1,"")
```

## 119. Location of First Number in a String

Suppose you have A1: = "abfg8ty#%473hj" and you want to find what is the position of first number in this. In this string, first number is 8 and its position is 5. You can use following formula –

```
=IFERROR(AGGREGATE(15,6,FIND({0,1,2,3,4,5,6,7,8,9},A1,SEQUENCE(LEN(A1))),1),"")
```

```
=IFERROR(AGGREGATE(15,6,FIND({0,1,2,3,4,5,6,7,8,9},A1,ROW(INDIRECT("1:"&LEN(A1))))),1,"")
```

## 120. Location of Last Number in a String

In the above example, last number is 3 and its position is 12. You can use following formula to find this –

```
=IFERROR(AGGREGATE(14,6,FIND({0,1,2,3,4,5,6,7,8,9},A1,SEQUENCE(LEN(A1))),1),"")
```

```
=IFERROR(AGGREGATE(14,6,FIND({0,1,2,3,4,5,6,7,8,9},A1,ROW(INDIRECT("1:"&LEN(A1))))),1,"")
```

## 121. Find the Value of First Non Blank Cell in a Range

```
=INDEX(FILTER(A1:A10,A1:A10<>0,""),1)
```

```
=IFERROR(INDEX(A1:A10,MATCH(TRUE,NOT(ISBLANK(A1:A10)),0)),"")
```

## 122. Find First Numeric Value in a Range

```
=INDEX(FILTER(A1:A100,ISNUMBER(A1:A100)),"",1)
```

```
=IFERROR(INDEX(A1:A100,MATCH(1,--ISNUMBER(A1:A100),0)),"")
```

## 123. Find Last Numeric Value in a Range

```
=IFERROR(1/LOOKUP(2,1/A1:A100),"")
```

## 124. Find First non Numeric Value in a Range

```
=INDEX(FILTER(A1:A100,ISTEXT(A1:A100)),1)
```

```
=IFERROR(INDEX(A1:A100,MATCH(1,--ISTEXT(A1:A100),0)), "")
```

## 125. Find Last non Numeric Value in a Range

```
=IFERROR(LOOKUP(REPT("z",255),A1:A100), "")
```

## 126. Find Last Used Value in a Range

```
= IFERROR(LOOKUP(2,1/(A1:A100<>""),A1:A100), "")
```

## 127. I have data for many years but I want the sum for only last 12 months

Let's assume that your data is aligned like this and you want to have sum only for last 12 months. This should take care of if future entries are done. Hence, if Jun data is entered, sum up has to start from Jun month. –

	A	B
1	Date	Sales
2	1/1/2013	1000
3	2/1/2013	4000
4	3/1/2013	4000
5	4/1/2013	4000
6	5/1/2013	3000
7	6/1/2013	8000
8	7/1/2013	9000
9	8/1/2013	7000
10	9/1/2013	8000
11	10/1/2013	9000
12	11/1/2013	3000
13	12/1/2013	3000
14	1/1/2014	6000
15	2/1/2014	1000
16	3/1/2014	7000
17	4/1/2014	9000
18	5/1/2014	2000

Use this formula –

```
=SUM(INDEX(SORTBY(B2:B20,A2:A20,-1),SEQUENCE(12)))
```

```
=IFERROR(SUM(OFFSET(INDIRECT("$B"&COUNTA($A:$A)),0,0,IF(COUNTA($A:$A)-12>0,-12,-COUNTA($A:$A)+1),1)), "")
```

Note – If you have only one column say B, replace \$A with \$B. (If only A, replace \$B with \$A)

## 128. Generate a Unique List out of Duplicate Entries

Suppose, you have entries in A2:A100 and you want to generate a list containing only unique entries in column C starting C2. You can use following formula in C2 and drag down the formula –

**Case 1 – A2:A100 doesn't contain any blanks**

=UNIQUE(A2:A100)

Older way -

=IFERROR(INDEX(\$A\$2:\$A\$100,MATCH(0,INDEX(COUNTIF(\$C\$1:\$C1,\$A\$2:\$A\$100),0,0)),),"")

**Case 2 – A2:A100 contains blanks**

=UNIQUE(FILTER(A2:A100,A2:A100<>""))

Older way - In this case, you will have to use Array formula.

=IFERROR(INDEX(\$A\$2:\$A\$100, MATCH(0, IF(\$A\$2:\$A\$100<>"",COUNTIF(C1:\$C\$1,\$A\$2:\$A\$100)), 0)),)"")

OR

=IFERROR(INDEX(\$A\$2:\$A\$100,MATCH(0,COUNTIF(\$C\$1:\$C1,\$A\$2:\$A\$100&""),0)),)"")

## 129. Financial Function - Calculate EMI

You want to take a loan and you want to calculate EMI OR you want to build an EMI calculator in Excel. It is a fairly easy job to do -

You will need to use PMT function for this. It has following structure -

PMT(rate, nper, pv, [fv], [type])

**rate:** You rate of interest

**nper:** No. of payments. Your nper and rate should be on the same scale. i.e if you are planning to pay up monthly, the rate in your formula should be monthly only. Generally, interest rate is specified yearly i.e. 10.5% per year. This you should divide by 12 to arrive at monthly rate. Hence, if you wanted 3 years loan, it means nper would  $3 \times 12 = 36$  months. If it is quarterly,  $\text{rate} = 10.5\% / 4 = 2.625\%$  and nper would be  $3 \times 4 = 12$ . If it is annual,  $\text{rate} = 10.5\%$  and nper = 3

**pv:** Your loan amount. You will need to put negative value of this in your formula. If you don't put negative value, your EMI would be in negative but answer would be same though with negative sign.

+ve / -ve PMT requires some explanation though you may choose to ignore. It depends upon your cashflow. If you are taking a loan, hence cash in, hence pv is +ve. But every month, you will have to pay up something, hence cash out. Hence, PMT is -ve. If you are investing, hence cash out. Hence pv is -ve. But every month, you will be receiving something, hence cash in. Hence, PMT is +ve.

Now what is +ve or -ve is simply your preference. I recommend you should not worry about this.

**fv:** Your remaining value after you finish your installment. Generally, it is 0 as any lender will like to recover its money full. (Default is 0)

**type:** 0 - At the end of the period, 1 - At the beginning of the period (Default is 0)  
Also note, fv and type are optional and may not be required in your formula.

The formula used in the below picture is =PMT(B1/12,B2,-B3,B4,B5)

**Bonus Tip** = If you use ABS function, then there would be no need to put negative value of PV. Hence, formula in this case would be =ABS(PMT(B1/12,B2,B3,B4,B5))

The below EMI Calculator can be download from here <http://eforexcel.com/wp/wp-content/uploads/2014/12/EMI-Calculator1.xlsx>

	A	B	C	D	E	F	G
1	Interest Rate	12%					
2	No. of Payments	36					
3	Loan Amount	10000					
4	Future Value	0					
5	EMI Due Type	0					
6							
7	EMI	\$332.14					

0 - At the end of the Period  
1 - At the beginning of the period

## 130. Financial Function - Calculate Interest Part of an EMI

Now the EMI for a month = Interest for that month and Principal for that month. IPMT is used to calculate the interest portion of your EMI.

Excel defines IPMT as "Returns the interest payment for a given period for an investment based on periodic, constant payments and a constant interest rate"

The syntax of IPMT is IPMT(rate, per, nper, pv, [fv], [type]).

**rate:** Your rate of interest

**per:** Period for which you want to calculate Interest

**nper:** No. of payments. Your nper and rate should be on the same scale. i.e if you are planning to pay up monthly, the rate in your formula should be monthly only. Generally, interest rate is specified yearly i.e. 10.5% per year. This you should divide by 12 to arrive at monthly rate. Hence, if you wanted 3 years loan, it means nper would  $3 \times 12 = 36$  months. If it is quarterly, rate =  $10.5\% / 4 = 2.625\%$  and nper would be  $3 \times 4 = 12$ . If it is annual, rate = 10.5% and nper = 3

**pv:** Your loan amount. You will need to put negative value of this in your formula. If you don't put negative value, your EMI would be in negative but answer would be same though with negative sign.

+ve / -ve IPMT requires some explanation though you may choose to ignore. It depends upon your cashflow. If you are taking a loan, hence cash in, hence pv is +ve. But every month, you will have to pay up something, hence cash out. Hence, IPMT is -ve. If you are investing, hence cash out. Hence pv is -ve. But every month, you will be receiving something, hence cash in. Hence, IPMT is +ve.

Now what is +ve or -ve is simply your preference. I recommend you should not worry about this.

**fv:** Your remaining value after you finish your installment. Generally, it is 0 as any lender will like to recover its money full. (Default is 0)

**type:** 0 - At the end of the period, 1 - At the beginning of the period (Default is 0)

Also note, fv and type are optional and may not be required in your formula.

The formula used in the below picture is `=IPMT(B1/12,B2,B3,-B4,B5,B6)`

Also, since Interest will vary every month, hence it makes sense to calculate it for each month. Columns H & I carry interest for each month.

**Bonus Tip** = If you use ABS function, then there would be no need to put negative value of PV. Hence, formula in this case would be `=ABS(IPMT(B1/12,B2,B3,B4,B5,B6))`

The below EMI Calculator can be download from here <http://eforexcel.com/wp/wp-content/uploads/2014/12/Interest-Payment-Calculator.xlsx>

	A	B	C	D	E	F	G	H	I
1	Interest Rate	12%						Period	Interest
2	Period for which Interest is wanted	24						1	\$100.00
3	No. of Payments	36						2	\$97.68
4	Loan Amount	10000						3	\$95.33
5	Future Value	0						4	\$92.97
6	EMI Due Type	0						5	\$90.57
7								6	\$88.16
8	<b>Interest</b>	<b>\$40.30</b>						7	\$85.72
9								8	\$83.25
10								9	\$80.77
11								10	\$78.25
12								11	\$75.71
13								12	\$73.15
14								13	\$70.56
15								14	\$67.94
16								15	\$65.30
17								16	\$62.63
18								17	\$59.94
19								18	\$57.21
20								19	\$54.47
21								20	\$51.69

## 131. Financial Function - Calculate Principal Part of an EMI

Now the EMI for a month = Interest for that month and Principal for that month. IPMT is used to calculate the interest portion of your EMI. To calculate the principal part of an EMI, you will need to use PPMT.

Excel defines PPMT as "Returns the payment on the principal for a given period for an investment based on periodic, constant payments and a constant interest rate."

The syntax of PPMT is PPMT(rate, per, nper, pv, [fv], [type])

**rate:** You rate of interest

**per:** Period for which you want to calculate Principal

**nper:** No. of payments. Your nper and rate should be on the same scale. i.e if you are planning to pay up monthly, the rate in your formula should be monthly only. Generally, interest rate is specified yearly i.e. 10.5% per year. This you should divide by 12 to arrive at monthly rate. Hence, if you wanted 3 years loan, it means nper would  $3 \times 12 = 36$  months. If it is quarterly, rate =  $10.5\% / 4 = 2.625\%$  and nper would be  $3 \times 4 = 12$ . If it is annual, rate = 10.5% and nper = 3

**pv:** Your loan amount. You will need to put negative value of this in your formula. If you don't put negative value, your EMI would be in negative but answer would be same though with negative sign.

+ve / -ve PPMT requires some explanation though you may choose to ignore. It depends upon your cashflow. If you are taking a loan, hence cash in, hence pv is +ve. But every month, you will have to pay up something, hence cash out. Hence, PPMT is -ve. If you are investing, hence cash out. Hence pv is -ve. But every month, you will be receiving something, hence cash in. Hence, PPMT is +ve.

Now what is +ve or -ve is simply your preference. I recommend you should not worry about this.

**fv:** Your remaining value after you finish your installment. Generally, it is 0 as any lender will like to recover its money fill. (Default is 0)

**type:** 0 - At the end of the period, 1 - At the beginning of the period (Default is 0)

Also note, fv and type are optional and may not be required in your formula.

The formula used in the below picture is =PPMT(B1/12,B2,B3,-B4,B5,B6)

Also, since Principal will vary every month, hence it makes sense to calculate it for each month. Columns H & I carry Principal for each month.

**Bonus Tip** = If you use ABS function, then there would be no need to put negative value of PV. Hence, formula in this case would be =ABS(PPMT(B1/12,B2,B3,B4,B5,B6))

The below EMI Calculator can be download from here <http://eforexcel.com/wp/wp-content/uploads/2014/12/Principal-Payment-Calculator.xlsx>

	A	B	C	D	E	F	G	H	I
1	Interest Rate		12%					Period	Principal
2	Period for which Principal is wanted		24					1	\$232.14
3	No. of Payments		36					2	\$234.46
4	Loan Amount		10000					3	\$236.81
5	Future Value		0					4	\$239.18
6	EMI Due Type		0					5	\$241.57
7								6	\$243.98
8	<b>Principal</b>		<b>\$291.84</b>					7	\$246.42
9								8	\$248.89
10								9	\$251.38
11								10	\$253.89
12								11	\$256.43
13								12	\$258.99
14								13	\$261.58
15								14	\$264.20
16								15	\$266.84
17								16	\$269.51
18								17	\$272.21
19								18	\$274.93
20								19	\$277.68
21								20	\$280.45

0 - At the end of the Period  
1 - At the beginning of the period

## 132. Financial Function - Calculate Number of EMIs to Pay Up a Loan

You have taken a loan and you know your EMI capability. So, you want to know how many months will be taken to pay off a loan completely.

It is fairly easy job to do it in Excel. You will need to use NPER function for this.

Excel defines NPER as "Returns the number of periods for an investment based on periodic, constant payments and a constant interest rate."



Syntax of NPER is `NPER(rate,pmt,pv,[fv],[type])`.

**rate:** You rate of interest

**pmt:** EMI (Payment per period). You will need to put -ve value of this in your formula. Your pmt and rate should be on the same scale. i.e if you are planning to pay up monthly, the rate in your formula should be monthly only. Generally, interest rate is specified yearly i.e. 10.5% per year. This you should divide by 12 to arrive at monthly rate. Hence, if you wanted 3 years loan, it means nper would  $3 \times 12 = 36$  months. If it is quarterly, rate =  $10.5\% / 4 = 2.625\%$  and nper would be  $3 \times 4 = 12$ . If it is annual, rate = 10.5% and nper = 3

**pv:** Your loan amount. You will need to put +ve value of this in your formula.  
Note - Either PMT or PV should be -ve. Both can't be +ve and -ve at the same time.

**+ve / -ve requires some explanation and this can not be ignored.** It depends upon your cashflow. If you are taking a loan, hence cash in, hence pv is +ve. But every month, you will have to pay up something, hence cash out. Hence, PMT is -ve. If you are investing, hence cash out. Hence pv is -ve. But every month, you will be receiving something, hence cash in. Hence, PMT is +ve.

**fv:** Your remaining value after you finish your installment. Generally, it is 0 as any lender will like to recover its money fill. (Default is 0)

**type:** 0 - At the end of the period, 1 - At the beginning of the period (Default is 0)

Also note, fv and type are optional and may not be required in your formula.

The formula used in the below picture is `=NPER(B1/12,-B2,B3)`

The below EMI Calculator can be download from here <http://eforexcel.com/wp/wp-content/uploads/2014/12/NPER-Calculator.xlsx>

	A	B	C	D	E	F	G
1	Interest Rate	12%					
2	EMI Per Month	332.14					
3	Loan Amount	10000					
4	Future Value	0					
5	EMI Due Type	0					
6							
7	<b>Payment Periods</b>	<b>36.00</b>					
8							
9							

0 - At the end of the Period  
 1 - At the beginning of the period

## 133. Financial Function - Calculate Interest Rate

You want to take a loan. You know how much loan to take (pmt), you know how many months you want to pay up (nper) and you want to know effective rate of interest. Excel makes it easy to do. RATE function is the answer for this.

Excel defines RATE as "Returns the interest rate per period of an annuity. RATE is calculated by iteration and can have zero or more solutions. If the successive results of RATE do not converge to within 0.0000001 after 20 iterations, RATE returns the #NUM! error value." Syntax of RATE is RATE(nper, pmt, pv, [fv], [type], [guess]).

**nper:** Payment periods. Typically in months.

**pmt:** EMI (Payment per period). You will need to put -ve value of this in your formula. Your pmt and rate should be on the same scale. i.e if you are planning to pay up monthly, the pmt in your formula should be monthly only.

**pv:** Your loan amount. You will need to put +ve value of this in your formula.  
Note - Either PMT or PV should be -ve. Both can't be +ve and -ve at the same time.

**+ve / -ve requires some explanation and this can not be ignored.** It depends upon your cashflow. If you are taking a loan, hence cash in, hence pv is +ve. But every month, you will have to pay up something, hence cash out. Hence, PMT is -ve. If you are investing, hence cash out. Hence pv is -ve. But every month, you will be receiving something, hence cash in. Hence, PMT is +ve.

**fv:** Your remaining value after you finish your installment. Generally, it is 0 as any lender will like to recover its money fill. (Default is 0)

**type:** 0 - At the end of the period, 1 - At the beginning of the period (Default is 0)

**guess:** If you omit guess, it is assumed to be 10 percent. If RATE does not converge, try different values for guess. RATE usually converges if guess is between 0 and 1. Once again, note that if PMT is monthly, then Guess should also be monthly. Hence, if you are giving annual interest rate of 12%, guess should be given as  $12\%/12 = 1\%$ .

Also note, fv, type and guess are optional and may not be required in your formula.

The formula used in the below picture is =RATE(B1,-B2,B3,B4,B5,B6/12)

The below EMI Calculator can be download from here <http://eforexcel.com/wp/wp-content/uploads/2014/12/RATE-Calculator.xlsx>

	A	B	C	D	E	F
1	Total Payment Periods		36 Months			
2	EMI Per Month		332.14			
3	Loan Amount		10000			
4	Future Value		0			
5	EMI Due Type		0			
6	Guess		9% Yearly			
7						
8	<b>Interest Rate</b>		<b>1.00%</b>			
9	<b>Annual Interest Rate</b>		<b>12.00%</b>			

0 - At the end of the Period  
 1 - At the beginning of the period

## 134. Financial Function – Calculate Compounded Interest

As part of our Mathematics courses in our childhood, we had learned about Compounded Interest. The famous formula which we remember is  
 Compounded Balance = Principal x (1+rate)^N

This is a fairly easy job to do in Excel. The formula to be used is FV.

Excel help describes FV as "Returns the future value of an investment based on periodic, constant payments and a constant interest rate".

The syntax of FV is FV(rate,nper,pmt,[pv],[type])

You require only 3 pieces of information for Compounded Balance.

**rate:** Interest rate on which compounding needs to be done

**nper:** Total number of periods for which compounding needs to be done. Now rate and nper should be on the same scale. If interest rate is monthly, then nper should be in months. If interest rate is quarterly, then nper should be in quarter. If interest rate is annual, then nper should be in years.

**pv:** This is the initial principal and it has to be specified in -ve. (Note, I have already discussed significance of +ve and -ve in many previous tips on Financial Functions.)

The formula used in below picture for Monthly

=FV(B1/12,B3\*12,0,-B2)

The formula used in below picture for Quarterly

=FV(F1/4,F3\*4,0,-F2)

The formula used in below picture for Yearly

=FV(J1,J3,0,-J2)

The Compounded Balance Calculator can be downloaded from <http://eforexcel.com/wp/wp-content/uploads/2014/12/Compounded-Interest-Calculator.xlsx>

	A	B	C	D	E	F	G	H	I	J	K
1	Interest	10%	Yearly		Interest	10%	Yearly		Interest	10%	Yearly
2	PV	1500			PV	1500			PV	1500	
3	NPER		5 Years		NPER		5 Years		NPER		5 Years
4											
5											
6	Balance After Compounding	\$2,467.96			Balance After Compounding	\$2,457.92			Balance After Compounding	\$2,415.77	
7	Compounded Interest	\$967.96			Compounded Interest	\$957.92			Compounded Interest	\$915.77	
8											
9	Monthly Compounding				Quarterly Compounding				Yearly Compounding		
10											

## 135. Financial Function – Calculate Effective Interest

You are applying for a loan and an interest rate has been quoted. The interest rate which is quoted is called "Nominal Interest Rate". They will quote Nominal Interest Rate in yearly terms. Hence, if they quote 12% interest for a loan, this is yearly figure. Now, you generally pay EMIs every month. They simply say that you need to pay 1% monthly interest which has been derived by annual interest rate / 12 which  $12\%/12=1\%$  in this case.

But actually interest rate of 1% (nominal monthly interest rate) is compounded every month, hence your effective interest rate per year becomes higher. But lending financial institutions doesn't quote this higher rate as it will make your loan cost look higher.

To calculate **Effective Interest Rate**, Excel has provided a function called EFFECT.

Excel describes EFFECT - Returns the effective annual interest rate, given the nominal annual interest rate and the number of compounding periods per year.

The syntax of EFFECT is EFFECT(nominal\_rate, npery).

**Nominal Rate** - Annual Interest Rate

**npery** - Compounding periods in a Year. For monthly payments, it is 12. For quarterly payments, it is 4.

In the below picture, the effective interest rate is 12.68% for a monthly payment. This may be a small difference for a year or two, but if you take mortgage on housing which is say for 20 years, this makes hell of a difference.

The formula used is =EFFECT(B1,B2)

Now, if you are making an investment and making monthly payments, you will be getting annual return of 12.68% against 12% if you make yearly payment.

The effective interest rate calculator can be downloaded from <http://eforexcel.com/wp/wp-content/uploads/2014/12/Effective-Interest-Calculator.xlsx>

	A	B	C	D	E	F	G
1	Interest	12%	Yearly		Interest	12%	Yearly
	Compounding				Compounding		
2	Periods in a Year	12	Months		Periods in a Year	4	Quarters
3							
4							
5	Effective Interest Rate	12.68%			Effective Interest Rate	12.55%	
6							
7	Monthly Compounding				Quarterly Compounding		
8							

## 136. Financial Function – Calculate CAGR and AAGR

Many times we get posed with the question of calculating CAGR (Compounded Annual Growth Rate) and AAGR (Average Annual Growth Rate). For layman's not knowing what are these, following are good resources to introduce people to CAGR and AAGR.

**CAGR** – <http://www.investopedia.com/terms/c/cagr.asp>

**AAGR** – <http://www.investopedia.com/terms/a/aagr.asp>

The formula to calculate CAGR is very simple. RATE function can be used to calculate CAGR. We need to know only PV, FV and NPER. Let's assume that you invested \$10,000 (hence, this is PV) and it has become \$16,448 (hence, this is FV) after the end of 5 years. So, to calculate CAGR, we will use following formula –

=RATE(5,-10000,16448)

=RATE(5,,10000,-16448)

Hence, in the above formulas, you need to put either PV as negative or FV as negative. Both can not be positive or negative at the same time.

If you don't have FV but a series of fixed payments (say investment of 3000 every year), you can use –

=RATE(5,-3000,10000)

=RATE(5,1000,-10000)

Now, let's come back to AAGR. There is no single formula for AAGR. If your data set is laid like below, you can use following formula to calculate AAGR –

=SUMPRODUCT((B3:B11-B2:B10)/(B2:B10))/(ROWS(B2:B10))

	A	B
1	Year	Amount
2	2005	10
3	2006	15
4	2007	18
5	2008	20
6	2009	22
7	2010	26
8	2011	29
9	2012	29
10	2013	30
11	2014	35

## 137. Slab Billing – Calculate Income Tax, Electricity (Utility) Bills based on Slabs

You will encounter slab billings in two very common documents – One is Electricity / Utility and another one is Income Tax. If you see your electricity bills, you will notice following type of entries (values are for illustration purposes only, please do not attach any meaning to them)

0 – 50 Units – \$1.5 per Unit  
 51-100 Units – \$ 2.0 per Unit  
 101-500 Units – \$3.5 Per Unit  
 501 – 2000 Units – \$6 Per Unit  
 2000 Unit Onwards – \$9 Per Unit

It means that your first 50 units will be charged at \$1.5 per unit, next 50 will be charged at \$2.0 per unit and next 400 units will be charged at \$3.5 per unit, next 1500 units will be charged at \$6 per unit and any unit after 2000 will be charged at \$ 9 per unit.

Hence, if your bill is 37 units, your charges will be =  $37 \times 1.5 = \$55.5$

Hence, if you bill is 87 units, you charges will be =  $50 \times 1.5 + (87 - 50) \times 2 = \$149$

Hence, if your bill is 415 units, your charges will be =  $50 \times 1.5 + 50 \times 2 + (415 - 100) \times 3.5 = \$1277.5$

Hence, if your bill is 780 units, your charges will be =  $50 \times 1.5 + 50 \times 2 + 400 \times 3.5 + (780 - 500) \times 6 = \$3255$

Hence, if your bill is 2670 units, your charges will be =  $50 \times 1.5 + 50 \times 2 + 400 \times 3.5 + 1500 \times 6 + (2670 - 2000) \times 9 = \$16605$

Same type of distribution, you will notice in Income Tax also. Consider the below slabs.

0 – \$10000 – Nil  
 \$10001 – \$50000 – 10%  
 \$50001 – \$100000 – 20%  
 \$100001 – \$500000 – 30%  
 \$500001 and above – 35%

Hence, if your income is \$8000, your income tax will be 0.

Hence, if your income is \$40000, your income tax will be  $= (40000 - 10000) * 10\% = \$3000$

Hence, if your income is \$90000, your income tax will be  $= 40000 * 10\% + (90000 - 50000) * 20\% = \$12000$

Hence, if your income is \$210000, your income tax will be  $= 40000 * 10\% + 50000 * 20\% + (210000 - 100001) * 30\% = \$47000$

Hence, if your income is \$850000, your income tax will be  $= 40000 * 10\% + 50000 * 20\% + 400000 * 30\% + (850000 - 500000) * 35\% = \$256500$

Download workbook related to this from <https://eforexcel.com/wp/wp-content/uploads/2015/12/Article-31-Slab-Billing.xlsx>

**Now, how to calculate this in Excel through a single formula.** Here, SUMPRODUCT comes to your rescue. The only per-requisite is that you need to set up your data appropriately for this purpose as given below.

	A	B	C	D	E	F
1	<b>From</b>	<b>To</b>	<b>Amount</b>			
2	0	50	1.5		<b>Units Consumed</b>	2670
3	50	100	2		<b>Bill</b>	16605
4	100	500	3.5			
5	500	2000	6			
6	2000	>2000	9			

**Then the formula for slab billing is simply following**

$=\text{SUMPRODUCT}((F2>A2:A6)*(F2-A2:A6)*(C2:C6-\text{IFERROR}(-C1:C5,0)))$

Below is the example for Income Tax

	J	K	L	M	N	O
1	<b>Slab</b>		<b>Amount</b>			
2	0	10000	0		<b>Income</b>	850000
3	10000	50000	10%		<b>Income Tax</b>	256500
4	50000	100000	20%			
5	100000	500000	30%			
6	500000	>500000	35%			

**Then the formula for Income Tax is simply following**

$=\text{SUMPRODUCT}((O2>J2:J6)*(O2-J2:J6)*(L2:L6-\text{IFERROR}(-L1:L5,0)))$

## 138. LTRIM and RTRIM through Excel Formulas

So, we have TRIM function in Excel. Almost all programming languages provide LTRIM and RTRIM functions also but Excel doesn't provide LTRIM and RTRIM. The same is provided in VBA but most of the Excel users are not using VBA. They are simple folks who want to accomplish their day to day job through Excel functions only.

Before I delve into LTRIM and RTRIM, let's recap TRIM.

TRIM, basically, removes all spaces from your string and if there are more than one space between your words, it will convert them into only one space. So, it removes all leading spaces, all trailing spaces and all spaces between words except one. If there is only one space between words, it will leave that space untouched.

Below are the examples of operating with TRIM function –

"Vijay A Verma" = "Vijay A Verma" (two spaces between Vijay and A have been reduced to one. Three spaces between A and Verma have been reduced to one)

" Vijay Verma" = "Vijay Verma"(Two spaces before Vijay have been reduced to no space and Six spaces between Vijay and Verma have been reduced to one)

"Vijay Verma " = "Vijay Verma"(Three spaces after Vijay have been reduced to no space and Six spaces between Vijay and Verma have been reduced to one)

" Vijay Verma " = "Vijay Verma" (Four spaces before Vijay have been reduced to no space, 3 spaces between Vijay and Verma have been reduced to one space and Eight spaces after Verma has been reduced to no space)

So, I am sure that now you have got what does TRIM do.

Now, I talked about that most programming languages including provide LTRIM and RTRIM function. So, what do LTRIM and RTRIM do –

1. LTRIM removes all leading spaces and leaves all other spaces untouched. Hence, if I operate LTRIM on following –

" Little Mary " = "Little Mary " (All four leading spaces have been trimmed so that is no space before first word. It has left spaces between two words and spaces after last word untouched)

2. RTRIM removes all trailing spaces and leaves all other spaces untouched. Hence, if I operate LTRIM on following –

" Little Mary " = " Little Mary" (All three trailing spaces have been trimmed so that is no space after last word. It has left spaces between two words and spaces before first word untouched)

Now, let's get back to making LTRIM and RTRIM through Excel functions-

The formula for LTRIM

=REPLACE(A1,1,FIND(LEFT(TRIM(A1),1),A1)-1,"")

So, this particular Excel formula would act like LTRIM that is it would remove all leading spaces and would leave all other spaces untouched.

The formula for RTRIM



```
=IFERROR(REPLACE(A1,MAX(INDEX((MID(A1,ROW(A1:INDEX(A:A,LEN(A1))),1)  
<>" ")*ROW(A1:INDEX(A:A,LEN(A1))),,))+1,LEN(A1),"",""))
```

Above function would act like RTRIM that is it would remove all trailing spaces and would leave all other spaces untouched.

---- End of Document ----